Completitud de la lógica proposicional

José A. Alonso Jiménez
Fabián Fernando Serrano Suárez

14 de octubre de 2009

Resumen

En este trabajo se presenta una formalización en Isabelle/HOL/Isar de un sistema axiomático de la lógica proposicional y la demostración de la completitud del sistema usando el método de Kalmar de eliminación de variables. La formalización se basa en la del libro de Mendelson ([1]). El objetivo es explorar la posibilidad de formalizar razonamiento complejo de forma legible por los humanos y procesable por las máquinas.

Índice

1 Un sistema de axiomas para el cálculo proposicional

1.1 Formalización del sistema axiomático

Definición 1.1 *El **alfabeto** de la lógica proposicional se compone de los siguientes símbolos:*

1. *Los **símbolos proposicionales**: $p_0, p_1, \ldots$.*
2. *Las **conectivas**: $\neg$ y $\rightarrow$.*
3. *Los **símbolos auxiliares**: "(" y ")".*

Definición 1.2 *Las **fórmulas** de la lógica proposicional se definen recursivamente por las siguientes reglas:*

1. *Los símbolos proposicionales son fórmulas.*
2. *Si F es una fórmula, entonces $\neg F$ también lo es.*
3. *Si F y G son fórmulas, entonces $(F \rightarrow G)$ también lo es.*

datatype *formula*
$=$ *Atom nat*
| *Neg formula* (*No* (-) [*110*] *110*)
| *Implica formula formula* (**infixl** $\rightarrow$ *101*)

Nota 1.3 *Usaremos los símbolos p, q y r en lugar de $p_0.p_1$ y p_2, respectivamente.*

constdefs
p :: *formula*
p $\equiv$ *Atom 0*
q :: *formula*
q $\equiv$ *Atom 1*
r :: *formula*
r $\equiv$ *Atom 2*

declare
p-def [*simp*]
q-def [*simp*]
r-def [*simp*]

Nota 1.4 *En la representación de las fórmulas se observa que:*

1. *El símbolo proposicional p_n se representa por Atom n.*
2. *La negación tiene precedencia sobre el condicional.*
3. *El condicional asocia por la izquierda.*

Con estos convenios, la expresión $(\neg\neg p \to q \to r)$ representa la fórmula $(((\neg(\neg p)) \to q) \to r)$.

lemma $(No\ No\ p \to q \to r) = (((No\ (No\ p)) \to q) \to r)$
by *simp*

Definición 1.5 *Los **axiomas** de la lógica proposicional son las fórmulas que tienen las siguientes formas:*

- $p \to (q \to p)$ *(A1)*
- $(p \to (q \to r)) \to ((p \to q) \to (p \to r))$ *(A2)*
- $(\neg q \to \neg p) \to ((\neg q \to p) \to q)$ *(A3)*

En las expresiones anteriores p, q y r pueden sustituirse por fórmulas cualesquiera.

constdefs
$A1 :: formula$
$A1 \equiv (p \to (q \to p))$

$A2 :: formula$
$A2 \equiv ((p \to (q \to r)) \to ((p \to q) \to (p \to r)))$

$A3 :: formula$
$A3 \equiv ((No\ q \to No\ p) \to ((No\ q \to p) \to q))$

$Axiomas :: formula\ set$
$Axiomas \equiv \{A1, A2, A3\}$

La siguiente declaración permite usar los axiomas como reglas de simplificación.

declare
A1-def [*simp*]
A2-def [*simp*]
A3-def [*simp*]
Axiomas-def [*simp*]

Definición 1.6 *Una **sustitución proposicional** es una aplicación de los números naturales en el conjunto de las fórmulas.*

types *sust* = *nat* ⇒ *formula*

Definición 1.7 *La **aplicación de la sustitución** ϑ a la fórmula F consiste en sustituir en F cada símbolo proposicional p_i por la fórmula $\vartheta(i)$.*

consts *sust* :: *sust* ⇒ *formula* ⇒ *formula*
primrec
sust ϑ (*Atom n*) = (ϑ *n*)
sust ϑ (*No F*) = *No* (*sust* ϑ *F*)
sust ϑ (*F1* → *F2*) = (*sust* ϑ *F1*) → (*sust* ϑ *F2*)

Definición 1.8 *La **sustitución vacía** aplica cada número natural i en la fórmula p_i. Se representa por ε.*

constdefs *vacia* :: *nat* ⇒ *formula* (ε)
vacia ≡ (λ*i*. (*Atom i*))

Ejemplo 1.9 *Al sustituir 0 por la fórmula F y 1 por la fórmula G en el axioma 1 se obtiene la fórmula $(F \to (G \to F))$. Se evalúa como sigue,*

normal-form *sust* (ε(*0*:=*F*,*1*:=*G*)) *A1*

Definición 1.10 *Sea S un conjunto de fórmulas. Una fórmula F es **deducible** a partir de S (y se representa por $S \vdash F$) si se obtiene aplicando las siguientes reglas:*

- *Si $F \in S$, entonces $S \vdash F$.*
- *Las fórmulas obtenidas aplicándole una sustitución proposicional a los axiomas son deducibles a partir de S.*

- *Si $F \to G$ y F son deducibles a partir de S, entonces G también lo es.*

inductive
esDeducible :: *formula set* ⇒ *formula* ⇒ *bool* (- ⊢ - [50,50] 50)
for S :: *formula set*
where
hip: ⟦ $F \in S$ ⟧ ⟹ $S \vdash F$
| *ax*: $F \in Axiomas \Longrightarrow S \vdash sust\ \vartheta\ F$
| *mp*: ⟦ $S \vdash (F \to G); S \vdash F$ ⟧ ⟹ $S \vdash G$

1.2 La regla de debilitamiento

Lema 1.11 (Debilitamiento) *Si $S \vdash F$ y $S \subseteq T$, entonces $T \vdash F$.*

Demostración: Por inducción sobre la deducibilidad de F. Se consideran tres casos:

Caso 1: Supongamos que $F \in S$. Entonces se prueba
$$\forall T.\ S \subseteq T \longrightarrow T \vdash F$$
usando la regla de las hipótesis.

Caso 2: Supongamos que F es un axioma. Entonces se prueba
$$\forall T.\ S \subseteq T \longrightarrow T \vdash \vartheta(F)$$
usando la regla de axiomas.

Caso 3: Supongamos que F se obtiene por modus ponens a partir de $G \to F$ y G. Por hipótesis de inducción se tiene
$$\forall T.\ S \subseteq T \longrightarrow T \vdash G \to F \quad (1)$$
$$\forall T.\ S \subseteq T \longrightarrow T \vdash G \quad (2)$$
y hay que demostrar que
$$\forall T.\ S \subseteq T \longrightarrow T \vdash F$$

Sea T tal que $S \subseteq T$. Por (1) y (2), se tiene
$$T \vdash G \to F \quad (3)$$
$$T \vdash G \quad (4)$$
Aplicando modus ponens a (3) y (4) se tiene
$$T \vdash F.$$

□

lemma *debilitamiento* [*rule-format*]:
$S \vdash F \Longrightarrow (\forall T.\ S \subseteq T \longrightarrow T \vdash F)$
proof (*induct rule*: *esDeducible.induct*)
— Caso 1:
{ **fix** F
assume $F \in S$

```
  thus ∀ T. S ⊆ T ⟶ T ⊢ F using hip by auto }
 — Caso 2:
 next
 { fix F ϑ
  assume F ∈ Axiomas
  thus ∀ T. S ⊆ T ⟶ T ⊢ sust ϑ F using ax by simp }
 — Caso 3:
 next
 { fix G F
  assume
    S ⊢ G → F and
    1: ∀ T. S ⊆ T ⟶ T ⊢ G → F and
    S ⊢ G and
    2: ∀ T. S ⊆ T ⟶ T ⊢ G
  show ∀ T. S ⊆ T ⟶ T ⊢ F
  proof
    fix T
    show S ⊆ T ⟶ T ⊢ F
    proof
      assume S ⊆ T
      hence T ⊢ G → F and T ⊢ G using 1 2 by auto
      thus T ⊢ F by (rule mp)
    qed
  qed }
qed
```

Corolario 1.12 (Segunda regla de debilitamiento) *Si $S \vdash F$, entonces $S \cup \{G\} \vdash F$.*

Demostración: Basta observar que $S \subseteq S \cup \{G\}$ y aplica la regla de debilitamiento. □

```
corollary debilitamiento2:
 assumes S ⊢ F
 shows S ∪ {G} ⊢ F
proof −
 have S ⊆ S ∪ {G} by auto
 with assms show S ∪ {G} ⊢ F by (rule debilitamiento)
qed
```

1.3 Teoremas elementales

Lema 1.13 (Axioma 1) $S \vdash F \to (G \to F)$.

Demostración: Por el axioma 1 y la regla de axiomas.

□

lemma *Axioma1*: $S \vdash F \to (G \to F)$
proof −
have $A1 \in Axiomas$ **by** *simp*
hence $S \vdash sust\ (\varepsilon(0{:=}F, 1{:=}G))\ A1$ **by** (*rule ax*)
thus *?thesis* **by** *simp*
qed

Lema 1.14 (Axioma 2) $S \vdash (F \to (G \to H)) \to ((F \to G) \to (F \to H))$.

Demostración: Por el axioma 2 y la regla de axiomas.

□

lemma *Axioma2*: $S \vdash (F \to (G \to H)) \to ((F \to G) \to (F \to H))$
proof −
have $A2 \in Axiomas$ **by** *simp*
hence $S \vdash sust\ (\varepsilon(0{:=}F, 1{:=}G, 2{:=}H))\ A2$ **by** (*rule ax*)
thus *?thesis* **by** *simp*
qed

Lema 1.15 (Axioma 3) $S \vdash (\neg G \to \neg F) \to ((\neg G \to F) \to G)$.

Demostración: Por el axioma 3 y la regla de axiomas.

□

lemma *Axioma3*: $S \vdash (No\ G \to No\ F) \to ((No\ G \to F) \to G)$
proof −
have $A3 \in Axiomas$ **by** *simp*
hence $S \vdash sust\ (\varepsilon(0{:=}F, 1{:=}G))\ A3$ **by** (*rule ax*)
thus *?thesis* **by** *simp*
qed

Lema 1.16 *Si $F \in S$, entonces $S \vdash G \to F$.*

Demostración:

1	$S \vdash F$	[hip]
2	$S \vdash F \to (G \to F)$	[Axioma 1]
3	$S \vdash G \to F$	[MP a 2 y 1]

□

lemma *por-hip*:

assumes $F \in S$
shows $S \vdash G \to F$
proof $-$
have *1*: $S \vdash F$ **using** *assms* **by** (*rule hip*)
have *2*: $S \vdash F \to (G \to F)$ **by** (*rule Axioma1*)
show $S \vdash G \to F$ **using** *2 1* **by** (*rule mp*)
qed

Lema 1.17 *Si $F \to G \in S$ y $F \in S$, entonces $S \vdash G$.*

Demostración: Supongamos que $F \to G \in S$ y $F \in S$, entonces (por hip) se tiene respectivamente que

$$S \vdash (F \to G) \tag{1}$$
$$S \vdash F \tag{2}$$

Aplicando MP a (1) y (2) se tiene que

$$S \vdash G.$$

□

lemma *mp-con-hip*:
assumes $F \to G \in S$
and $F \in S$
shows $S \vdash G$
proof $-$
have *1*: $S \vdash (F \to G)$ **using** *assms*(*1*) **by** (*rule hip*)
have *2*: $S \vdash F$ **using** *assms*(*2*) **by** (*rule hip*)
show $S \vdash G$ **using** *1 2* **by** (*rule mp*)
qed

Lema 1.18 (Identidad [1] p. 36) $S \vdash F \to F$.

Demostración:

1.	$S \vdash (F \to ((F \to F) \to F)) \to ((F \to (F \to F)) \to (F \to F))$	[Axioma 2]
2.	$S \vdash F \to ((F \to F) \to F)$	[Axioma 1]
3.	$S \vdash (F \to (F \to F)) \to (F \to F)$	[MP a 1 y 2]
4.	$S \vdash F \to (F \to F)$	[Axioma 1]
5.	$S \vdash F \to F$	[MP a 3 y 4]

□

lemma *identidad*: $S \vdash F \to F$
proof $-$
have *1*: $S \vdash (F \to ((F \to F) \to F)) \to ((F \to (F \to F)) \to (F \to F))$ **by** (*rule Axioma2*)
have *2*: $S \vdash F \to ((F \to F) \to F)$ **by** (*rule Axioma1*)

have *3*: $S \vdash (F \to (F \to F)) \to (F \to F)$ **using** *1 2* **by** (*rule mp*)
have *4*: $S \vdash F \to (F \to F)$ **by** (*rule Axioma1*)
show *5*: $S \vdash F \to F$ **using** *3 4* **by** (*rule mp*)
qed

1.4 Teorema de la deducción

Lema 1.19 (Lema de deducción ([1] p. 37)) *Si $S \vdash F$, entonces para toda G se tiene que si $S = \{G\} \cup S'$, entonces $S' \vdash G \to F$.*

Demostración: Por inducción sobre la deducibilidad de F. Se consideran tres casos:

Caso 1: Supongamos que $F \in S$. Hay que probar que
$$\forall\, G\, S'.\ S = \{G\} \cup S' \longrightarrow S' \vdash G \to F$$

Sean G y S' tales que $S = \{G\} \cup S'$. Entonces $F \in \{G\} \cup S'$. Pueden darse dos casos

Caso 1a: Supongamos que $F \in \{G\}$. Entonces, $F = G$ y se tiene que $S' \vdash G \to F$ por el lema 1.18.

Caso 1b: Supongamos que $F \in S'$. Entonces se tiene que $S' \vdash G \to F$ por el lema 1.16.

Caso 2: Supongamos que F es un axioma. Hay que probar que
$$\forall\, G\, S'.\ S = \{G\} \cup S' \longrightarrow S' \vdash G \to \mathit{sust}\ \vartheta\ F$$

Sean G y S' tales que $S = \{G\} \cup S'$. Entonces

1	$S' \vdash \mathit{sust}\ \vartheta\ F \to (G \to \mathit{sust}\ \vartheta\ F)$	[Axioma 1]
2	$S' \vdash \mathit{sust}\ \vartheta\ F$	[ax y F es axioma]
3	$S' \vdash G \to \mathit{sust}\ \vartheta\ F$	[MP a 1 y 2]

Caso 3: Supongamos que F se obtiene por modus ponens a partir de $G \to F$ y G. Por hipótesis de inducción se tiene que
$$\forall\, H\, S'.\ S = \{H\} \cup S' \longrightarrow S' \vdash H \to (G \to F) \tag{5}$$
$$\forall\, H\, S'.\ S = \{H\} \cup S' \longrightarrow S' \vdash H \to G \tag{6}$$
y hay que demostrar que
$$\forall\, H\, S'.\ S = \{H\} \cup S' \longrightarrow S' \vdash H \to F$$

Sean H y S' tales que
$$S = \{H\} \cup S' \tag{7}$$
Entonces,

8	$S' \vdash H \to (G \to F)$	[de 5 y 7]
9	$S' \vdash H \to G$	[de 6 y 7]
10	$S' \vdash (H \to (G \to F)) \to ((H \to G) \to (H \to F))$	[Axioma 2]
11	$S' \vdash (H \to G) \to (H \to F)$	[MP 10 y 8]
12	$S' \vdash H \to F$	[MP 11 y 9]

□

lemma *lema-deduccion*:
$S \vdash F \Longrightarrow (\forall\, G\, S'.\, S = \{G\} \cup S' \longrightarrow S' \vdash (G \to F))$
proof(*induct rule*: *esDeducible.induct*)
— Caso 1:
{ **fix** F
assume $F \in S$
show $\forall\, G\, S'.\, S = \{G\} \cup S' \longrightarrow S' \vdash G \to F$
proof
fix G
show $\forall\, S'.\, S = \{G\} \cup S' \longrightarrow S' \vdash G \to F$
proof
fix S'
show $S = \{G\} \cup S' \longrightarrow S' \vdash G \to F$
proof
assume $S = \{G\} \cup S'$
show $S' \vdash G \to F$
proof (*rule UnE*)
show $F \in \{G\} \cup S'$ **using** ‹$F \in S$› **and** ‹$S = \{G\} \cup S'$› **by** *simp*
next
— Caso 1a:
{ **assume** $F \in \{G\}$
thus $S' \vdash G \to F$ **by** (*simp add*:*identidad*) }
next
— Caso 1b:
{ **assume** $F \in S'$
thus $S' \vdash G \to F$ **by** (*simp add*:*por-hip*) }
qed
qed
qed
qed }
next
— Caso 2:
{ **fix** $F\ \vartheta$
assume $F \in Axiomas$
show $\forall\, G\, S'.\, S = \{G\} \cup S' \longrightarrow S' \vdash G \to sust\ \vartheta\ F$
proof
fix G
show $\forall\, S'.\, S = \{G\} \cup S' \longrightarrow S' \vdash G \to sust\ \vartheta\ F$
proof
fix S'
show $S = \{G\} \cup S' \longrightarrow S' \vdash G \to sust\ \vartheta\ F$
proof

```
      assume S = {G} ∪ S'
      show S' ⊢ G → sust ϑ F
      proof −
       have 1: S' ⊢ sust ϑ F → (G → sust ϑ F) by (rule Axioma1)
       have 2: S' ⊢ sust ϑ F using ⟨F ∈ Axiomas⟩ by (rule ax)
       show S' ⊢ G → sust ϑ F using 1 2 by (rule mp)
      qed
     qed
    qed
   qed }
 next
 — Caso 3:
 { fix G F
  assume
    S ⊢ G → F and
    5: ∀ H S'. S = {H} ∪ S' ⟶ S' ⊢ H → (G → F) and
    S ⊢ G and
    6: ∀ H S'. S = {H} ∪ S' ⟶ S' ⊢ H → G
  show ∀ H S'. S = {H} ∪ S' ⟶ S' ⊢ H → F
  proof
   fix H
   show ∀ S'. S = {H} ∪ S' ⟶ S' ⊢ H → F
   proof
    fix S'
    show S = {H} ∪ S' ⟶ S' ⊢ H → F
    proof
     assume 7: S = {H} ∪ S'
     show S' ⊢ H → F
     proof −
      have 8: S' ⊢ H → (G → F) using 5 7 by simp
      have 9: S' ⊢ H → G using 6 7 by simp
      have S' ⊢ (H → (G → F)) → ((H → G) → (H → F)) by (rule Axioma2)
      hence S' ⊢ (H → G) → (H → F) using 8 by (rule mp)
      thus S' ⊢ H → F using 9 by (rule mp)
     qed
    qed
   qed
  qed }
qed
```

Teorema 1.20 (Teorema de deducción ([1] p. 37))
Si $\{F\} \cup S \vdash G$, *entonces* $S \vdash (F \to G)$.

Demostración: Es consecuencia inmediata del lema de deducción.

□

theorem *deduccion*:
$(\{F\} \cup S) \vdash G \Longrightarrow S \vdash (F \to G)$
by (*simp add*:*lema-deduccion*)

1.5 Algunas consecuencias del teorema de la deducción

Lema 1.21 (Regla de corte) *Si* $S \vdash F$ *y* $(\{F\} \cup S) \vdash G$, *entonces* $S \vdash G$.

Demostración: De la segunda hipótesis y el teorema de la deducción, tenemos que $S \vdash (F \to G)$. De esto último y la primera hipótesis, usando MP, tenemos que $S \vdash G$.

□

lemma *regla-de-corte*:
assumes
$S \vdash F$ **and**
$(\{F\} \cup S) \vdash G$
shows $S \vdash G$
proof −
have $S \vdash (F \to G)$ **using** *assms*(2) **by** (*rule deduccion*)
thus $S \vdash G$ **using** *assms*(1) **by** (*rule mp*)
qed

Lema 1.22 (Encadenamiento ([1] p. 38)) $\{F \to G, G \to H\} \vdash F \to H$.

Demostración:

1.	$\{F \to G, G \to H, F\} \vdash F \to G$	[hipótesis]
2.	$\{F \to G, G \to H, F\} \vdash G \to H$	[hipótesis]
3.	$\{F \to G, G \to H, F\} \vdash F$	[hipótesis]
4.	$\{F \to G, G \to H, F\} \vdash G$	[MP, 1 y 3]
5.	$\{F \to G, G \to H, F\} \vdash H$	[MP, 2 y 4]
6.	$\{F \to G, G \to H\} \vdash (F \to H)$	[Teorema deducción, 5]

□

lemma *encadenamiento*: $\{F \to G, G \to H\} \vdash F \to H$
proof −
have *1*: $\{F \to G, G \to H, F\} \vdash F \to G$ **using** *hip* **by** *simp*
have *2*: $\{F \to G, G \to H, F\} \vdash G \to H$ **using** *hip* **by** *simp*
have *3*: $\{F \to G, G \to H, F\} \vdash F$ **using** *hip* **by** *simp*
have *4*: $\{F \to G, G \to H, F\} \vdash G$ **using** *1 3* **by** (*rule mp*)
have *5*: $\{F \to G, G \to H, F\} \vdash H$ **using** *2 4* **by** (*rule mp*)

show *6*: $\{F \to G, G \to H\} \vdash (F \to H)$ **using** *5* **by** (*simp add:deduccion*)
qed

Lema 1.23 (Segunda regla de encadenamiento)
Si $S \vdash F \to G$ *y* $S \vdash G \to H$, *entonces* $S \vdash F \to H$.

Demostración:

1	$S \cup \{F\} \vdash F$	[Hip]
2	$S \cup \{F\} \vdash F \to G$	[Debilitamiento2, H1]
3	$S \cup \{F\} \vdash G$	[MP, 2, 1]
4	$S \cup \{F\} \vdash G \to H$	[Debilitamiento2, H2]
5	$S \cup \{F\} \vdash H$	[MP, 4, 3]
6	$S \vdash F \to H$	[Deduccion, 5]

□

lemma *encadenamiento2*:
assumes
$S \vdash F \to G$
$S \vdash G \to H$
shows
$S \vdash F \to H$
proof −
have *1*: $S \cup \{F\} \vdash F$ **by** (*simp add:hip*)
have *2*: $S \cup \{F\} \vdash F \to G$ **using** *assms*(*1*) **by** (*rule debilitamiento2*)
have *3*: $S \cup \{F\} \vdash G$ **using** *2 1* **by** (*rule mp*)
have *4*: $S \cup \{F\} \vdash G \to H$ **using** *assms*(*2*) **by** (*rule debilitamiento2*)
have *5*: $S \cup \{F\} \vdash H$ **using** *4 3* **by** (*rule mp*)
thus *6*: $S \vdash F \to H$ **by** (*simp add:deduccion*)
qed

Lema 1.24 (MP2 ([1] p. 38)) $\{F \to (G \to H), G\} \vdash (F \to H)$.

Demostración:

1	$\{F \to (G \to H), G, F\} \vdash F \to (G \to H)$	[hipótesis]
2	$\{F \to (G \to H), G, F\} \vdash G$	[hipótesis]
3	$\{F \to (G \to H), G, F\} \vdash F$	[hipótesis]
4	$\{F \to (G \to H), G, F\} \vdash G \to H$	[MP, 1 y 3]
5	$\{F \to (G \to H), G, F\} \vdash H$	[MP, 4 y 2]
6	$\{F \to (G \to H), G\} \vdash (F \to H)$	[Teorema deducción, 5]

□

lemma *mp2*: $\{F \to (G \to H), G\} \vdash (F \to H)$

proof −
have *1*: $\{F \to (G \to H), G, F\} \vdash F \to (G \to H)$ **using** *hip* **by** *simp*
have *2*: $\{F \to (G \to H), G, F\} \vdash G$ **using** *hip* **by** *simp*
have *3*: $\{F \to (G \to H), G, F\} \vdash F$ **using** *hip* **by** *simp*
have *4*: $\{F \to (G \to H), G, F\} \vdash G \to H$ **using** *1 3* **by** (*rule mp*)
have *5*: $\{F \to (G \to H), G, F\} \vdash H$ **using** *4 2* **by** (*rule mp*)
show *6*: $\{F \to (G \to H), G\} \vdash (F \to H)$ **using** *5* **by** (*simp add*:*deduccion*)
qed

Lema 1.25 (MP2a) *Si* $S \vdash F \to (G \to H)$ *y* $S \vdash G$, *entonces* $S \vdash F \to H$.

Demostración:

1	$S \cup \{F\} \vdash F$	[Hip]
2	$S \cup \{F\} \vdash F \to (G \to H)$	[Debilitamiento2, H1]
3	$S \cup \{F\} \vdash G \to H$	[MP, 2, 1]
4	$S \cup \{F\} \vdash G$	[Debilitamiento2, H2]
5	$S \cup \{F\} \vdash H$	[MP, 3, 4]
6	$S \vdash F \to H$	[Deduccion, 5]

□

lemma *mp2a*:
assumes
$S \vdash F \to (G \to H)$
$S \vdash G$
shows
$S \vdash F \to H$
proof −
have *1*: $S \cup \{F\} \vdash F$ **by** (*simp add*:*hip*)
have *2*: $S \cup \{F\} \vdash F \to (G \to H)$ **using** *assms*(*1*) **by** (*rule debilitamiento2*)
have *3*: $S \cup \{F\} \vdash G \to H$ **using** *2 1* **by** (*rule mp*)
have *4*: $S \cup \{F\} \vdash G$ **using** *assms*(*2*) **by** (*rule debilitamiento2*)
have *5*: $S \cup \{F\} \vdash H$ **using** *3 4* **by** (*rule mp*)
thus *6*: $S \vdash F \to H$ **by** (*simp add*:*deduccion*)
qed

Lema 1.26 (Segunda regla de corte) *Si* $S \vdash F$, $S \vdash G$ *y* $\{F, G\} \vdash H$, *entonces* $S \vdash H$.

Demostración: Puesto que $S \subseteq \{G\} \cup S$, de la hipótesis $S \vdash F$, por la regla de debilitamiento, se tiene que

$$(\{G\} \cup S) \vdash F \tag{1}$$

De igual forma, puesto que $\{F, G\} \subseteq \{F\} \cup (\{G\} \cup S)$, de la hipótesis $\{F, G\} \vdash H$, por la regla de debilitamiento, se tiene que

$$(\{F\} \cup (\{G\} \cup S)) \vdash H \tag{2}$$

De (1) y (2), por el lema 1.21, obtenemos

$$(\{G\} \cup S) \vdash H \tag{3}$$

De lo anterior y de la hipótesis $S \vdash G$, por el lema 1.21, se demuestra que $S \vdash H$.

□

lemma *corte2*:
assumes $S \vdash F$ **and** $S \vdash G$ **and** $\{F, G\} \vdash H$
shows $S \vdash H$
proof −
have $S \subseteq \{G\} \cup S$ **by** *auto*
with *assms*(*1*) **have** *1*: $(\{G\} \cup S) \vdash F$ **by** (*rule debilitamiento*)
have $\{F, G\} \subseteq \{F\} \cup (\{G\} \cup S)$ **by** *auto*
with *assms*(*3*) **have** *2*: $(\{F\} \cup (\{G\} \cup S)) \vdash H$ **by** (*rule debilitamiento*)
have *3*: $(\{G\} \cup S) \vdash H$ **using** *1 2* **by** (*rule regla-de-corte*)
with *assms*(*2*) **show** $S \vdash H$ **by** (*rule regla-de-corte*)
qed

Lema 1.27 (Eliminación de la doble negación ([1] p. 38)) $S \vdash \neg\neg F \to F$.

Demostración:

1	$S \vdash (\neg F \to \neg\neg F) \to ((\neg F \to \neg F) \to F)$	[Axioma3]
2	$S \vdash \neg F \to \neg F$	[Identidad]
3	$S \vdash (\neg F \to \neg\neg F) \to F$	[MP, 1, 2]
4	$S \vdash \neg\neg F \to (\neg F \to \neg\neg F)$	[Axioma1]
5	$S \vdash \neg\neg F \to F$	[Encadenamiento2, 4, 3]

□

lemma *eliminaDN*: $S \vdash No\ No\ F \to F$
proof −
have *1*: $S \vdash (No\ F \to No\ No\ F) \to ((No\ F \to No\ F) \to F)$ **by** (*rule Axioma3*)
have *2*: $S \vdash No\ F \to No\ F$ **by** (*rule identidad*)
have *3*: $S \vdash (No\ F \to No\ No\ F) \to F$ **using** *1 2* **by** (*rule mp2a*)
have *4*: $S \vdash No\ No\ F \to (No\ F \to No\ No\ F)$ **by** (*rule Axioma1*)
show *5*: $S \vdash No\ No\ F \to F$ **using** *4 3* **by** (*rule encadenamiento2*)
qed

Lema 1.28 (Introducción de la doble negación ([1] p. 36)) $S \vdash F \to \neg\neg F$.

Demostración:

1	$S \vdash (\neg\neg\neg F \to \neg F) \to ((\neg\neg\neg F \to F) \to \neg\neg F)$	[Axioma3]
2	$S \vdash \neg\neg\neg F \to \neg F$	[EliminaDN]
3	$S \vdash (\neg\neg\neg F \to F) \to \neg\neg F$	[MP 1, 2]
4	$S \vdash F \to (\neg\neg\neg F \to F)$	[Axioma1]
5	$S \vdash F \to \neg\neg F$	[Encadenamiento2 4, 3]

□

lemma *introDN*: $S \vdash F \to No\ No\ F$
proof −
have *1*: $S \vdash (No\ No\ No\ F \to No\ F) \to ((No\ No\ No\ F \to F) \to No\ No\ F)$
by (*rule Axioma3*)
have *2*: $S \vdash No\ No\ No\ F \to No\ F$ **by** (*rule eliminaDN*)
have *3*: $S \vdash (No\ No\ No\ F \to F) \to No\ No\ F$ **using** *1 2* **by** (*rule mp*)
have *4*: $S \vdash F \to (No\ No\ No\ F \to F)$ **by** (*rule Axioma1*)
show *5*: $S \vdash F \to No\ No\ F$ **using** *4 3* **by** (*rule encadenamiento2*)
qed

Lema 1.29 (Elimación de la negación ([1] p. 36)) $S \vdash \neg F \to (F \to G)$.

Demostración:

1	$\{\neg F, F\} \cup S \vdash \neg F$	[Hipótesis]
2	$\{\neg F, F\} \cup S \vdash F$	[Hipótesis]
3	$\{\neg F, F\} \cup S \vdash F \to (\neg G \to F)$	[Axioma1]
4	$\{\neg F, F\} \cup S \vdash \neg F \to (\neg G \to \neg F)$	[Axioma1]
5	$\{\neg F, F\} \cup S \vdash \neg G \to F$	[MP 3,2]
6	$\{\neg F, F\} \cup S \vdash \neg G \to \neg F$	[MP 4,1]
7	$\{\neg F, F\} \cup S \vdash (\neg G \to \neg F) \to ((\neg G \to F) \to G)$	[Axioma3]
8	$\{\neg F, F\} \cup S \vdash (\neg G \to F) \to G$	[MP 7,6]
9	$\{\neg F, F\} \cup S \vdash G$	[MP 8,5]
10	$\{\neg F\} \cup S \vdash F \to G$	[Deducción 9]
11	$S \vdash \neg F \to (F \to G)$	[Deduccion 10]

□

lemma *elimNeg*: $S \vdash No\ F \to (F \to G)$
proof −
have *1*: $\{No\ F, F\} \cup S \vdash No\ F$ **using** *hip* **by** *simp*
have *2*: $\{No\ F, F\} \cup S \vdash F$ **using** *hip* **by** *simp*
have *3*: $\{No\ F, F\} \cup S \vdash F \to (No\ G \to F)$ **by** (*rule Axioma1*)
have *4*: $\{No\ F, F\} \cup S \vdash No\ F \to (No\ G \to No\ F)$ **by** (*rule Axioma1*)
have *5*: $\{No\ F, F\} \cup S \vdash No\ G \to F$ **using** *3 2* **by** (*rule mp*)
have *6*: $\{No\ F, F\} \cup S \vdash No\ G \to No\ F$ **using** *4 1* **by** (*rule mp*)
have *7*: $\{No\ F, F\} \cup S \vdash (No\ G \to No\ F) \to ((No\ G \to F) \to G)$ **by** (*rule Axioma3*)
have *8*: $\{No\ F, F\} \cup S \vdash (No\ G \to F) \to G$ **using** *7 6* **by** (*rule mp*)

have *9*: $\{No\ F, F\} \cup S \vdash G$ **using** *8 5* **by** (*rule mp*)
have *10*: $\{No\ F\} \cup S \vdash F \to G$ **using** *9* **by** (*simp add*: *deduccion*)
show *11*: $S \vdash No\ F \to (F \to G)$ **using** *10* **by** (*rule deduccion*)
qed

Lema 1.30 (Contrarecíproco ([1] p. 36)) $S \vdash (\neg G \to \neg F) \to (F \to G)$.

Demostración:

1	$\{\neg G \to \neg F\} \cup S \vdash \neg G \to \neg F$	[Hip]
2	$\{\neg G \to \neg F\} \cup S \vdash (\neg G \to \neg F) \to ((\neg G \to F) \to G)$	[Axioma3]
3	$\{\neg G \to \neg F\} \cup S \vdash F \to (\neg G \to F)$	[Axioma1]
4	$\{\neg G \to \neg F\} \cup S \vdash (\neg G \to F) \to G$	[MP,2,1]
5	$\{\neg G \to \neg F\} \cup S \vdash F \to G$	[Encadenamiento2,3,4]
6	$S \vdash (\neg G \to \neg F) \to (F \to G)$	[Deduccion,5]

□

lemma *contrareciproco-1*: $S \vdash (No\ G \to No\ F) \to (F \to G)$
proof −
have *1*: $\{No\ G \to No\ F\} \cup S \vdash No\ G \to No\ F$ **by** (*simp add*:*hip*)
have *2*: $\{No\ G \to No\ F\} \cup S \vdash (No\ G \to No\ F) \to ((No\ G \to F) \to G)$
by (*rule Axioma3*)
have *3*: $\{No\ G \to No\ F\} \cup S \vdash F \to (No\ G \to F)$ **by** (*rule Axioma1*)
have *4*: $\{No\ G \to No\ F\} \cup S \vdash (No\ G \to F) \to G$ **using** *2 1* **by** (*rule mp*)
have *5*: $\{No\ G \to No\ F\} \cup S \vdash F \to G$ **using** *3 4* **by** (*rule encadenamiento2*)
thus *6*: $S \vdash (No\ G \to No\ F) \to (F \to G)$ **by** (*rule deduccion*)
qed

Lema 1.31 (Contrarecíproco ([1] p. 36)) $S \vdash (F \to G) \to (\neg G \to \neg F)$.

Demostración:

1	$\{F \to G\} \cup S \vdash F \to G$	[Hip]
2	$\{F \to G\} \cup S \vdash \neg\neg F \to F$	[EliminaDN]
3	$\{F \to G\} \cup S \vdash \neg\neg F \to G$	[Encadenamiento2,2,1]
4	$\{F \to G\} \cup S \vdash G \to \neg\neg G$	[IntroDN]
5	$\{F \to G\} \cup S \vdash \neg\neg F \to \neg\neg G$	[Encadenamiento2,3,4]
6	$\{F \to G\} \cup S \vdash (\neg\neg F \to \neg\neg G) \to (\neg G \to \neg F)$	[Contrareciproco 1]
7	$\{F \to G\} \cup S \vdash \neg G \to \neg F$	[MP,6,5]
8	$S \vdash (F \to G) \to (\neg G \to \neg F)$	[Deduccion,7]

□

lemma *contrareciproco-2*: $S \vdash (F \to G) \to (No\ G \to No\ F)$
proof −

have *1*: $\{F \to G\} \cup S \vdash F \to G$ **by** (*simp add:hip*)
have *2*: $\{F \to G\} \cup S \vdash No\ No\ F \to F$ **by** (*rule eliminaDN*)
have *3*: $\{F \to G\} \cup S \vdash No\ No\ F \to G$ **using** *2 1* **by** (*rule encadenamiento2*)
have *4*: $\{F \to G\} \cup S \vdash G \to No\ No\ G$ **by** (*rule introDN*)
have *5*: $\{F \to G\} \cup S \vdash No\ No\ F \to No\ No\ G$ **using** *3 4* **by** (*rule encadenamiento2*)
have *6*: $\{F \to G\} \cup S \vdash (No\ No\ F \to No\ No\ G) \to (No\ G \to No\ F)$ **by** (*rule contrareciproco-1*)
have *7*: $\{F \to G\} \cup S \vdash No\ G \to No\ F$ **using** *6 5* **by** (*rule mp*)
thus *8*: $S \vdash (F \to G) \to (No\ G \to No\ F)$ **by** (*rule deduccion*)
qed

Lema 1.32 (Negación del condicional ([1] p. 36)) $S \vdash F \to (\neg G \to \neg(F \to G))$.

Demostración:

1	$\{F \to G, F\} \cup S \vdash F \to G$	[Hip]
2	$\{F \to G, F\} \cup S \vdash F$	[Hip]
3	$\{F \to G, F\} \cup S \vdash G$	[MP, 1, 2]
4	$\{F\} \cup S \vdash (F \to G) \to G$	[Deducción, 3]
5	$S \vdash F \to ((F \to G) \to G)$	[Deducción, 4]
6	$S \vdash ((F \to G) \to G) \to (\neg G \to \neg(F \to G))$	[Contrarecíproco 2]
7	$S \vdash F \to (\neg G \to \neg(F \to G))$	[Encadenamiento2, 5, 6]

□

lemma *negacion-del-condicional*: $S \vdash F \to (No\ G \to No\ (F \to G))$
proof −
have *1*: $\{F \to G, F\} \cup S \vdash F \to G$ **by** (*simp add:hip*)
have *2*: $\{F \to G, F\} \cup S \vdash F$ **by** (*simp add:hip*)
have *3*: $\{F \to G, F\} \cup S \vdash G$ **using** *1 2* **by** (*rule mp*)
hence *4*: $\{F\} \cup S \vdash (F \to G) \to G$ **using** *deduccion* **by** *auto*
hence *5*: $S \vdash F \to ((F \to G) \to G)$ **by** (*rule deduccion*)
have *6*: $S \vdash ((F \to G) \to G) \to (No\ G \to No\ (F \to G))$ **by** (*rule contrareciproco-2*)
show *7*: $S \vdash F \to (No\ G \to No\ (F \to G))$ **using** *5 6* **by** (*rule encadenamiento2*)
qed

Lema 1.33 (Uso del tercio excluso ([1] p. 36)) $S \vdash (F \to G) \to ((\neg F \to G) \to G)$.

Demostración:

1	$\{F \to G, \neg F \to G\} \cup S \vdash F \to G$	[Hip]
2	$\{F \to G, \neg F \to G\} \cup S \vdash \neg F \to G$	[Hip]
3	$\{F \to G, \neg F \to G\} \cup S \vdash (F \to G) \to (\neg G \to \neg F)$	[Contrareciproco_2]
4	$\{F \to G, \neg F \to G\} \cup S \vdash \neg G \to \neg F$	[MP, 3, 1]
5	$\{F \to G, \neg F \to G\} \cup S \vdash (\neg F \to G) \to (\neg G \to \neg\neg F)$	[Contrareciproco_2]
6	$\{F \to G, \neg F \to G\} \cup S \vdash \neg G \to \neg\neg F$	[MP, 5, 2]
7	$\{F \to G, \neg F \to G\} \cup S \vdash (\neg G \to \neg\neg F) \to ((\neg G \to \neg F) \to G)$	[Axioma3]
8	$\{F \to G, \neg F \to G\} \cup S \vdash (\neg G \to \neg F) \to G$	[MP, 7, 6]
9	$\{F \to G, \neg F \to G\} \cup S \vdash G$	[MP, 8, 4]
10	$\{F \to G\} \cup S \vdash (\neg F \to G) \to G$	[Deducción, 9]
11	$S \vdash (F \to G) \to ((\neg F \to G) \to G)$	[Deducción, 10]

□

lemma *tercio-excluso-1*: $S \vdash (F \to G) \to ((No\ F \to G) \to G)$
proof −
have *1*: $\{F \to G, No\ F \to G\} \cup S \vdash F \to G$ **by** (*simp add*:*hip*)
have *2*: $\{F \to G, No\ F \to G\} \cup S \vdash No\ F \to G$ **by** (*simp add*:*hip*)
have *3*: $\{F \to G, No\ F \to G\} \cup S \vdash (F \to G) \to (No\ G \to No\ F)$
by (*rule contrareciproco-2*)
have *4*: $\{F \to G, No\ F \to G\} \cup S \vdash No\ G \to No\ F$ **using** *3 1* **by** (*rule mp*)
have *5*: $\{F \to G, No\ F \to G\} \cup S \vdash (No\ F \to G) \to (No\ G \to No\ No\ F)$
by (*rule contrareciproco-2*)
have *6*: $\{F \to G, No\ F \to G\} \cup S \vdash No\ G \to No\ No\ F$ **using** *5 2* **by** (*rule mp*)
have *7*: $\{F \to G, No\ F \to G\} \cup S \vdash (No\ G \to No\ No\ F) \to ((No\ G \to No\ F) \to G)$
by (*rule Axioma3*)
have *8*: $\{F \to G, No\ F \to G\} \cup S \vdash (No\ G \to No\ F) \to G$ **using** *7 6* **by** (*rule mp*)
have *9*: $\{F \to G, No\ F \to G\} \cup S \vdash G$ **using** *8 4* **by** (*rule mp*)
have *10*: $\{F \to G\} \cup S \vdash (No\ F \to G) \to G$ **using** *9* **by** (*simp add*: *deduccion*)
show *11*: $S \vdash (F \to G) \to ((No\ F \to G) \to G)$ **using** *10* **by** (*rule deduccion*)
qed

Lema 1.34 (Tercio excluso) *Si* $(\{F\} \cup S) \vdash G$ *y* $(\{\neg F\} \cup S) \vdash G$, *entonces* $S \vdash G$.

Demostración:

1	$S \vdash F \to G$	[Deducción, H1]
2	$S \vdash \neg F \to G$	[Deducción, H2]
3	$S \vdash (F \to G) \to ((\neg F \to G) \to G)$	[Tercio_excluso_1]
4	$S \vdash (\neg F \to G) \to G$	[MP, 3, 1]
5	$S \vdash G$	[MP, 4, 2]

□

lemma *tercio-excluso*:
assumes $(\{F\} \cup S) \vdash G$ **and** $(\{No\ F\} \cup S) \vdash G$

shows $S \vdash G$
proof $-$
have *1*: $S \vdash F \rightarrow G$ **using** *assms*(*1*) **by** (*rule deduccion*)
have *2*: $S \vdash No\ F \rightarrow G$ **using** *assms*(*2*) **by** (*rule deduccion*)
have *3*: $S \vdash (F \rightarrow G) \rightarrow ((No\ F \rightarrow G) \rightarrow G)$ **by** (*rule tercio-excluso-1*)
have *4*: $S \vdash (No\ F \rightarrow G) \rightarrow G$ **using** *3 1* **by** (*rule mp*)
show *5*: $S \vdash G$ **using** *4 2* **by** (*rule mp*)
qed

Corolario 1.35 *Sea S un conjunto finito. Si* $(\{F\} \cup S) \vdash G$ *y* $(\{\neg F\} \cup S) \vdash G$*, entonces* $S \vdash G$*.*

corollary *tercio-excluso-2*:
assumes *insert F S* $\vdash G$ **and** *insert* $(No\ F)\ S \vdash G$
shows $S \vdash G$
using *tercio-excluso*[*of F S G*] *assms* **by** *simp*

2 Semántica de la lógica proposicional

Definición 2.1 *Los* ***valores de verdad*** *son* V *(que se interpreta como* ***verdadero****) y* F *(que se interpreta como* ***falso****).*

En la siguiente formalización se identifica el conjunto de los valores de verdad con el tipo *v-verdad*, el valor V con la constante *Verdad* y el valor F con la constante *Falso* respectivamente.

datatype *v-verdad* $=$ *Verdad* | *Falso*

Definición 2.2 *Una* ***interpretación*** *es una aplicación del conjunto de los índices de los símbolos proposicionales en el conjunto de los valores de verdad.*

types *interpretacion* $=$ *nat* $\Rightarrow$ *v-verdad*

Definición 2.3 *El* ***valor*** *de una fórmula F en una interpretación I (representado por* $I'(F)$*) se define por recursión como sigue:*

- $I'(p_n) = I(n)$ *si* p_n *es un símbolo proposicional*
- $I'(\neg F) = \begin{cases} \mathsf{V}, & si\ I'(F) = \mathsf{F} \\ \mathsf{F}, & si\ I'(F) = \mathsf{V} \end{cases}$

- $I'(F \to G) = \begin{cases} \mathsf{F}, & \textit{si } I'(F) = \mathsf{V} \textit{ y } I'(G) = \mathsf{F} \\ \mathsf{V}, & \textit{en caso contrario.} \end{cases}$

En lo sucesivo, escribiremos $I(F)$ en lugar de $I'(F)$.

constdefs *v-negacion* :: *v-verdad* ⇒ *v-verdad*
v-negacion x == (*if x* = *Verdad then Falso else Verdad*)

constdefs *v-implicacion* :: *v-verdad* ⇒ *v-verdad* ⇒ *v-verdad*
v-implicacion x y == (*if x* = *Falso then Verdad else y*)

consts *valor* :: *interpretacion* ⇒ *formula* ⇒ *v-verdad*
primrec
valor I (*Atom n*) = *I n*
valor I (*No F*) = (*v-negacion* (*valor I F*))
valor I (*F* → *G*) = (*v-implicacion* (*valor I F*) (*valor I G*))

Definición 2.4 *Una fórmula F es una* ***tautología*** *si $I(F) = \mathsf{V}$ para toda interpretación I.*

constdefs *tautologia* :: *formula* ⇒ *bool*
tautologia F ≡ (∀ *I*. ((*valor I F*) = *Verdad*))

Las fórmulas que tienen la forma de axioma son tautologías.

Lema 2.5 (tautología-A1) *La fórmula a1*: $F \to (G \to H)$ *es tautología.*

Demostración: Sea I una interpretación, hay que demostrar que $I(a1) = \mathsf{V}$. Como se muestra en la siguiente tabla, la demostración es por casos en los valores posibles de F en I y utilizando la definición del valor de una implicación . El símbolo - significa valores arbitrarios.

	F G	$F \to (G \to F)$
caso 1	V -	V V V
caso 2	F -	F V

□

lemma *tautologia-A1*: *tautologia* (*F* → (*G* → *F*))
proof −
have ∀ *I*. *valor I* (*F* → (*G* → *F*)) = *Verdad*
proof
fix *I*

```
  show valor I (F → (G → F)) = Verdad
  proof (cases valor I F)
   — Caso 1:
   { assume valor I F = Verdad
   thus ?thesis by (simp add: v-implicacion-def) }
   next
   — Caso 2:
   { assume valor I F = Falso
   thus ?thesis by(simp add: v-implicacion-def) }
  qed
 qed
 thus ?thesis by (simp add: tautologia-def)
qed
```

Lema 2.6 (tautología-A2) *La fórmula a2*: $(F \to (G \to H)) \to ((F \to G) \to (F \to H))$ *es tautología.*

Demostración: Sea I una interpretación, hay que demostrar que $I(a2) = \mathsf{V}$. Como se muestra en la siguiente tabla, la demostración es por casos en los valores posibles de F,H y G, respectivamente, en I y utilizando la definición del valor de una implicación.

	F H G	$(F \to (G \to H)) \to ((F \to G) \to (F \to H))$
caso 1a	V V -	V V V V
caso 1b.1	V F V	V F V F F V
caso 1b.2	V F F	V V F F V
caso 2	F - -	V V F V

□

```
lemma tautologia-A2: tautologia ((F → (G → H)) → ((F → G) → (F → H)))
proof −
 have ∀ I. valor I ((F → (G → H)) → ((F → G) → (F → H))) = Verdad
 proof
  fix I
  show valor I ((F → (G → H)) → ((F → G) → (F → H))) = Verdad
  proof (cases valor I F)
   — Caso 1:
   { assume valor I F = Verdad
   show ?thesis
   proof (cases valor I H)
    — Caso 1a:
    { assume valor I H = Verdad
    thus ?thesis by (simp add: v-implicacion-def) }
```

```
    next
    — Caso 1b
    { assume Vr: valor I H = Falso
    show ?thesis
    proof (cases valor I G)
     — Caso 1b.1:
     { assume valor I G = Verdad
     thus ?thesis using Vr by (simp add: v-implicacion-def) }
     next
     — Caso 1b.2:
     { assume valor I G = Falso
     thus ?thesis by (simp add: v-implicacion-def) }
    qed }
   qed  }
   next
   — Caso 2:
   { assume  valor I F = Falso
   thus ?thesis by (simp add: v-implicacion-def) }
  qed
 qed
 thus ?thesis   by (simp add: tautologia-def)
qed
```

Lema 2.7 (tautología-A3) *La fórmula a3*: $(\neg G \rightarrow \neg F) \rightarrow ((\neg G \rightarrow F) \rightarrow G)$ *es tautología.*

Demostración: Sea I una interpretación, hay que demostrar que $I(a3) = \mathsf{V}$. Como se muestra en la siguiente tabla, la demostración es por casos en los valores posibles de G y F, respectivamente, en I y utilizando la definiciónes del valor de una implicación y una negación.

	G	F	$(\neg G \rightarrow \neg F) \rightarrow ((\neg G \rightarrow F) \rightarrow G)$
caso 1	V	-	V V V
caso 2a	F	V	V F F V
caso 2b	F	F	V V F F V

□

```
lemma tautologia-A3: tautologia ((No G → No F) → ((No G → F) → G))
proof −
 have ∀ I. valor I ((No G → No F) → ((No G → F) → G)) = Verdad
 proof
  fix I
  show valor I ((No G → No F) → ((No G → F) → G)) = Verdad
  proof (cases valor I G)
```

```
    — Caso 1:
    { assume valor I G = Verdad
    thus ?thesis by (simp add: v-implicacion-def) }
    next
    — Caso 2:
    { assume valor I G = Falso
    show ?thesis
    proof (cases valor I F)
     — Caso 2a:
     { assume valor I F = Verdad
     thus ?thesis by (simp add: v-negacion-def)(simp add: v-implicacion-def) }
     next
     — Caso 2b:
     { assume valor I F = Falso
     thus ?thesis by (simp add: v-negacion-def)(simp add: v-implicacion-def) }
    qed }
   qed
  qed
  thus ?thesis by (simp add: tautologia-def)
qed
```

Definición 2.8 *Una interpretación I es **modelo** de un conjunto de fórmulas S si, para toda fórmula F de S, $I(F) = \mathsf{V}$.*

```
constdefs modelo :: interpretacion ⇒ formula set ⇒ bool (- mod - [80,80] 80)
  I mod S ≡ (∀ F ∈ S. valor I F = Verdad)
```

Definición 2.9 *Un conjunto de fórmulas es **satisfactible** si tiene algún modelo. En caso contrario se dice que es **insatisfactible**.*

```
constdefs satisfactible :: formula set ⇒ bool
  satisfactible S ≡ (∃ v. v mod S)
```

Definición 2.10 *Una fórmula F es una **consecuencia lógica** de un conjunto de fórmulas S si para todo modelo I de S se tiene que $I(F) = \mathsf{V}$. Se representa por $S \models F$.*

```
constdefs consecuencia :: formula set ⇒ formula ⇒ bool (- ⊨ - [80,80] 80)
  S ⊨ F ≡ (∀ I. I mod S ⟶ valor I F = Verdad)
```

En particular, la fórmula F es una tautología si $\emptyset \models F$.

Lema 2.11 *Todas las interpretaciones son modelo del conjunto vacío.*

Demostración: Por la definición de modelo de un conjunto.

□

lemma *modelo-de-vacio*:
$\forall I.\ I\ mod\ \{\}$
proof −
have $\forall F \in \{\}.\ valor\ I\ F = Verdad$ **by** *simp*
thus $\forall I.\ I\ mod\ \{\}$ **by** (*simp add*: *modelo-def*)
qed

Lema 2.12 *Una fórmula es tautología syss es consecuencia del conjunto vacío.*

Demostración: Es consecuencia del lema anterior y de las definiciones de consecuencia lógica y de tautología.

□

lemma *CNS-tautologia*:
$tautologia\ F = (\{\} \models F)$
by (*simp add*: *tautologia-def consecuencia-def modelo-de-vacio*)

Lema 2.13 (CNS-tautología1) *Si una fórmula es tautología, es consecuencia lógica de cualquier conjunto.*

Demostración: Basta con aplicar las definciones de tautología y consecuencia lógica.

□

lemma *CNS-tautologia1*:
assumes *tautologia F*
shows $S \models F$
proof −
show *?thesis* **using** *assms* **by** (*simp add*: *tautologia-def consecuencia-def*)
qed

3 Teorema de validez

Teorema 3.1 (de validez ([1] p. 40)) *Las fórmulas deducibles a partir de un conjunto S de fórmulas son consecuencias de S; es decir, si* $S \vdash F$*, entonces* $S \models F$*.*

Demostración: Por inducción sobre la deducibilidad de F. Se consideran tres casos:

Caso 1: Supongamos que $F \in S$. Sea I un modelo de S, entonces, por definición de modelo, $I(G) = \mathsf{V}$ para toda fórmula $G \in S$. Luego, por hipótesis, $I(F) = \mathsf{V}$. De esta forma, $S \models F$.

Caso 2: Supongamos que F es un axioma y ϑ es una sustitución. Entonces F es uno de los axiomas *A1*, *A2* o *A3*, por lo tanto, por los lemas 2.5, 2.6, 2.7 se tiene que *sust* ϑ F es una tautología. Luego, por 2.13, $S \vdash$ *sust* ϑ F.

Caso 3: Supongamos que F se obtiene por modus ponens a partir de $G \to F$ y G. Por hipótesis de inducción se tiene que $S \models F \to G$ y $S \models F$. Luego, por definicion de consecuencia lógica, para todo modelo I de S se tiene que $I(F \to G) = \mathsf{V}$ y $I(F) = \mathsf{V}$. Por lo tanto, por la definición del valor de una implicación, necesariamente se tiene que $I(G) = \mathsf{V}$. Así, queda demostrado que $S \models G$.

□

theorem *validez*: $S \vdash F \implies S \models F$
proof (*induct rule*: *esDeducible.induct*)
— Caso 1:
{**fix** F
assume *hip1*: $F \in S$
have $\forall I.\ I$ *mod* $S \longrightarrow$ *valor* $I\ F = Verdad$
proof
fix I
show
I *mod* $S \longrightarrow$ *valor* $I\ F = Verdad$
proof
assume
I *mod* S
thus *valor* $I\ F = Verdad$ **using** *hip1* **by** (*simp add*: *modelo-def*)
qed
qed
thus $S \models F$ **by** (*simp add*: *consecuencia-def*)
}
next
— Caso 2:
{**fix** $F\ \vartheta$
assume *hip2*: $F \in Axiomas$
have $S \models$ *sust* ϑ *A1* **using** *tautologia-A1* **and** *CNS-tautologia1* **by** *simp*
moreover
have $S \models$ *sust* ϑ *A2* **using** *tautologia-A2* **and** *CNS-tautologia1* **by** *simp*
moreover
have $S \models$ *sust* ϑ *A3* **using** *tautologia-A3* **and** *CNS-tautologia1* **by** *simp*
ultimately

show $S \models$ *sust* ϑ F **using** *hip2* **by** *auto*
}
next
— Caso 3:
{**fix** F G
assume
$S \vdash F \to G$ **and**
1: $S \models F \to G$ **and**
$S \vdash F$ **and**
2: $S \models F$
show $S \models G$ **using** *1* **and** *2* **by** (*simp add*: *consecuencia-def v-implicacion-def*)
}
qed

4 Teorema de completitud

En esta sección se formaliza en Isar la demostración del teorema de completitud, el cual afirma que en nuestro sistema deductivo se pueden deducir *todas* las tautologías. Un sistema con esta propiedad se dice que es *deductivamente completo*.

4.1 Prueba informal

En esta parte presentamos una prueba del teorema de completitud, similar a las que aparecen en los textos clásicos de lógica, por ejemplo en [1] p. 42.

Definición 4.1 *Sea I una interpretación. La **relativización** de una fórmula F respecto de I se define por*

$$F^I = \begin{cases} F, & si\ I(F) = \mathsf{V} \\ \neg F, & si\ I(F) = \mathsf{F} \end{cases}$$

Esta definición nos permite enunciar el siguiente lema a partir del cual se demuestra el teorema de completitud.

Lema 4.2 (Kalmar ([1] p. 41)) *Sean $p_1, p_2, \ldots, p_m$ los símbolos proposicionales de F y sea I una interpretación, entonces $\{p_1^I, \ldots, p_m^I\} \vdash F^I$.*

Demostración: La prueba es por inducción en el número n de ocurrencias de conectivas en F.

Caso base. Supongamos que $n = 0$. Entonces F es un símbolo proposicional p_1, luego para cada uno de los casos $I(p_1) = \mathsf{V}$ e $I(p_1) = \mathsf{F}$ el lema $\{p_1^I\} \vdash p_1^I$ se reduce a las afirmaciones $\{p_1\} \vdash p_1$ y $\{\neg p_1\} \vdash \neg p_1$ respectivamente, que son verdaderas por la definición 1.10.

Paso de inducción. Supongamos que el lema se tiene para $j < n$. Distinguimos los siguientes casos.

Caso 1. Supongamos que F es $\neg G$. Entonces G tiene menos de n ocurrencias de conectivas, y tiene los mismos símbolos proposicionales que F. Luego, por hipótesis de inducción, se tiene que

$$\{p_1^I, \ldots, p_m^I\} \vdash G^I. \tag{1}$$

Distinguimos los siguientes dos subcasos.

Caso 1a. Supongamos que $I(G) = \mathsf{V}$, entonces $I(F) = \mathsf{F}$. Por lo tanto, $G^I = G$ y $F^I = \neg F$. Por consiguiente,

1. $\{p_1^I, \ldots, p_m^I\} \vdash G$ [por (1) y $G^I = G$]
2. $\{p_1^I, \ldots, p_m^I\} \vdash G \to (\neg\neg G)$ [lema 1.28]
3. $\{p_1^I, \ldots, p_m^I\} \vdash \neg\neg G$ [MP 2 y 1]
4. $\{p_1^I, \ldots, p_m^I\} \vdash \neg F$ [F es $\neg G$]
5. $\{p_1^I, \ldots, p_m^I\} \vdash F^I$ [$F^I = \neg F$]

Caso 1b. Supongamos que $I(G) = \mathsf{F}$, entonces $I(F) = \mathsf{V}$. Por lo tanto, $G^I = \neg G$ y $F^I = F$. Por la hipótesis de inducción tenemos que, $\{p_1^I, \ldots, p_m^I\} \vdash \neg G$. De esto último, y teniendo en cuenta que $\neg G$ es F^I, se tiene el lema.

Caso 2. Supongamos que F es $G_1 \to G_2$. Entonces, G_1 y G_2 tienen menos ocurrencias de conectivas que F. Luego, por hipótesis de inducción $S \vdash G_1^I$ y $T \vdash G_2^I$, en donde S y T son los conjuntos de símbolos proposicionales que ocurren en G_1 y G_2 respectivamente. Luego,

$$\{p_1^I, \ldots, p_m^I\} \vdash G_2^I \tag{2a}$$
$$\{p_1^I, \ldots, p_m^I\} \vdash G_1^I \tag{2b}$$

Distinguimos los siguientes tres subcasos.

Caso 2a. Supongamos que $I(G_2) = \mathsf{V}$, entonces $I(F) = \mathsf{V}$. Luego, $G_2^I = G_2$ y $F^I = F$. Por lo tanto,

1. $\{p_1^I, \ldots, p_m^I\} \vdash G_2$ [por (2a)]
2. $\{p_1^I, \ldots, p_m^I\} \vdash G2 \to (G1 \to G2)$ [por axioma A1]
3. $\{p_1^I, \ldots, p_m^I\} \vdash G_1 \to G_2$ [por MP 2 y 1]
4. $\{p_1^I, \ldots, p_m^I\} \vdash F$ [F es $G_1 \to G_2$]
5. $\{p_1^I, \ldots, p_m^I\} \vdash F^I$ [$F^I = F$]

Caso 2b. Supongamos que $I(G_1) = \mathsf{V}$ y $I(G_2) = \mathsf{V}$. Entonces $I(F) = \mathsf{F}$. Luego, $G_1^I = G_1$, $G_2^I = \neg G_2$, y $F^I = \neg F$. Por lo tanto,

1. $\{p_1^I, \ldots, p_m^I\} \vdash G_1$ [por (2b) y $G_1^I = G_1$]
2. $\{p_1^I, \ldots, p_m^I\} \vdash \neg G_2$ [por (2a) y $G_2^I = \neg G_2$]
3. $\{p_1^I, \ldots, p_m^I\} \vdash G1 \to (\neg G2 \to \neg(G_1 \to G_2))$ [por lema 1.32]
4. $\{p_1^I, \ldots, p_m^I\} \vdash \neg(G_1 \to G_2)$ [MP 3 y 1]
5. $\{p_1^I, \ldots, p_m^I\} \vdash \neg F$ [4 y F es $G_1 \to G_2$]
6. $\{p_1^I, \ldots, p_m^I\} \vdash F^I$ [5 y $F^I = \neg F$]

Caso 2c. Supongamos que $I(G_1) = \mathsf{F}$. Entonces $I(F) = \mathsf{V}$. Luego, $G_1^I = \neg G_1$ y $F^I = F$. Por lo tanto,

1. $\{p_1^I, \ldots, p_m^I\} \vdash \neg G_1$ [por (2a) y $G_1^I = \neg G_1$]
2. $\{p_1^I, \ldots, p_m^I\} \vdash \neg G_1 \to (G_1 \to G_2)$ [por el lema 1.29]
3. $\{p_1^I, \ldots, p_m^I\} \vdash G_1 \to G_2$ [MP 2 y 1]
4. $\{p_1^I, \ldots, p_m^I\} \vdash F$ [3 y F es $G_1 \to G_2$]
5. $\{p_1^I, \ldots, p_m^I\} \vdash F^I$ [4 y $F^I = F$]

□

Teorema 4.3 (Teorema de completitud) *Si* $\vDash F$ *entonces* $\vdash F$.

Demostración: Sea F una tautología. Entonces, para toda interpretación I, se tiene que $I(F) = \mathsf{V}$ y, por lo tanto, $F^I = F$. Por el lema 4.2 tenemos que

$$\{p_1^I, \ldots, p_n^I\} \vdash F \tag{1}$$

Sea I una interpretación de $p_2, \ldots, p_n$. Entonces I se puede extender a $p_1, p_2, \ldots, p_n$ de dos maneras: $I(p_1) = \mathsf{V}$ en cuyo caso $p_1^I = p_1$ y tenemos por (1)

$$\{p_1, p_2^I, \ldots, p_n^I\} \vdash F,$$

o $I(p_1) = \mathsf{F}$ en cuyo caso $p_1^I = \neg p_1$ y por lo tanto

$$\{\neg p_1, p_2^I, \ldots, p_n^I\} \vdash F.$$

Así, por el teorema del medio excluso, tenemos que

$$\{p_2^I, \ldots, p_n^I\} \vdash F.$$

Si repetimos este proceso n veces podemos eliminar todas las premisas hasta obtener $\{\} \vdash F$.

□

Obsérvese que en la última argumentación de la prueba anterior aparece la parte informal de la demostración. Para su formalización se debe tener un procedimiento que permita la eliminación de hipótesis (premisas). En lo que sigue hacemos una nueva presentación del lema de Kalmar con base a los conceptos de ”hipótesis” y ”conclusión” de una fórmula, a partir de los cuales se demuestra un método de eliminación de hipótesis de un conjunto de símbolos proposicionales en la deducción de una fórmula.

4.2 Prueba formal

Definición 4.4 *Sea T un conjunto de índices de símbolos proposicionales. El conjunto de* ***las hipótesis de una fórmula*** *F* ***con respecto de*** *T se define de la manera siguiente.*

1. $hip_T(p_n) = \begin{cases} \{p_n\}, & si\ n \in T \\ \{\neg p_n\}, & en\ caso\ contrario \end{cases}$
2. $hip_T(\neg G) = hip_T(G)$
3. $hip_T(G \to H) = hip_T(G) \cup hip_T(H)$

```
primrec hip :: nat set ⇒ formula ⇒ formula set where
 hip T (Atom n) = (if n ∈ T then {Atom n} else {No (Atom n)})
| hip T (No  F) = hip T F
| hip T (F → G) = hip T F ∪ hip T G
```

Algunas propiedades de los conjuntos de hipótesis son las siguientes.

Lema 4.5 *El conjunto $hip_T(F)$ es finito.*

Demostración: La demostración es por inducción en la fórmula F.

Caso 1. Supongamos que F es un símbolo proposicional. Entonces, por definición, el número de elementos de $hip_T(F)$ es igual a 1.

Caso 2. Supongamos que $hip_T(F)$ es finito, entonces $hip_T(\neg F)$ es finito ya que por definición $hip_T(F) = hip_T(\neg F)$.

Caso 3. Supongamos que $hip_T(F)$ y $hip_T(G)$ son finitos. Entonces, $hip_T(F \to G)$ es finito ya que por definición $hip_T(F \to G) = hip_T(F) \cup hip_T(G)$ y la unión de conjuntos finitos es finito.

□

```
lemma hip-finito: finite (hip T F)
proof (induct F)
 — Caso 1:
 { fix nat
 show  finite (hip T (Atom nat)) by simp }
 next
 — Caso 2:
 { fix F
 assume finite (hip T F)
 show finite (hip T (No F)) by simp }
```

next
— Caso 3:
{ **fix** *F G*
assume *finite* (*hip T F*) **and** *finite* (*hip T G*)
show *finite* (*hip T* (*F* → *G*)) **by** *auto* }
qed

La demostración del lema anterior se puede hacer de forma más automática:

lemma *hip-finito1*: *finite* (*hip T F*)
proof −
show *?thesis* **by** (*rule induct*) *auto*
qed

En forma análoga se tiene la demostración de los siguientes lemas

Lema 4.6 *Todo elemento de* $hip_T(F)$ *es un símbolo proposicional que está en T o es la negación de un símbolo proposicional que no pertenece a T.*

Demostración: Por inducción en la fórmula *F*.

□

lemma *pertenece-hip* [*rule-format*]:
G ∈ *hip T F* ⟶
(∃ *n*. (*G* = *Atom n* ∧ *n* ∈ *T*) ∨ (*G* = *No* (*Atom n*) ∧ *n* ∉ *T*))
proof −
show *?thesis* **by** (*rule induct*) *auto*
qed

Lema 4.7 *El conjunto de las hipótesis de una fórmula F respecto del conjunto de índices* $T - \{n\}$ *está contenido en la unión de* $\{\neg p_n\}$ *con el conjunto de hipótesis de F respecto de T distintas de* p_n*:*

$$hip_{T-\{n\}}(F) \subseteq \{\neg p_n\} \cup (hip_T(F) - \{p_n\}).$$

Demostración: Por inducción en la fórmula *F*.

□

lemma *hip-diferencia*: *hip* (*T*−{*n*}) *F* ⊆ {*No* (*Atom n*)}∪ ((*hip T F*) − {*Atom n*})
proof −
show *?thesis* **by** (*rule induct*) *auto*
qed

Lema 4.8 *El conjunto de las hipótesis de una fórmula F respecto del conjunto de índices* $\{n\} \cup T$ *está contenido en la unión de* $\{p_n\}$ *con el conjunto de hipótesis de F respecto de T distintas de* $\neg p_n$*:*

$$hip_{\{n\}\cup T}(F) \subseteq \{p_n\} \cup (hips_T(F) - \{\neg p_n\}).$$

Demostración: Por inducción en la fórmula F.

□

lemma *hip-cons*:
hip $(\{n\} \cup T)$ $F \subseteq \{Atom\ n\} \cup ((hip\ T\ F) - \{No\ (Atom\ n)\})$
proof −
show *?thesis* **by** (*rule induct*) *auto*
qed

Definición 4.9 *Sea T un conjunto de índices de símbolos proposicionales.*

La ***interpretación asociada*** *a T es la interpretación definida por*

$$v_T(p) = \begin{cases} \mathsf{V}, & si\ p \in T \\ \mathsf{F}, & de\ lo\ contrario. \end{cases}$$

La ***conclusión de la fórmula*** *F* ***con respecto de*** *T es*

$$con_T(F) = \begin{cases} F, & si\ v_T(F) = \mathsf{V} \\ \neg F, & si\ v_T(F) = \mathsf{F} \end{cases}$$

definition *interpretacion-asociada* :: $(nat\ set) \Rightarrow (nat \Rightarrow v\text{-}verdad)$ **where**
interpretacion-asociada T $\equiv \lambda\ x.$ (*if x* $\in$ *T then Verdad else Falso*)

definition *con* :: $(nat\ set) \Rightarrow formula \Rightarrow formula$ **where**
con T F $\equiv$ (*if valor* (*interpretacion-asociada T*) *F* = *Verdad then F else No F*)

Los siguientes lemas son consecuencia directa de los conceptos anteriores.

Lema 4.10 *Sea T un conjunto de índices de símbolos proposicionales. El valor de la conclusión de F respecto de T en la interpretación asociada a T es verdadero.*

lemma *valor-con-interpretacion-asociada*:
valor (*interpretacion-asociada T*)(*con T F*) = *Verdad*
proof(*cases valor* (*interpretacion-asociada T*) *F*)
assume *hip*: *valor* (*interpretacion-asociada T*) *F* = *Verdad*

```
 hence con T F = F by (simp add: con-def)
 thus ?thesis using hip by simp
 next
 assume hip: valor (interpretacion-asociada T) F = Falso
 hence con T F = (No F) by (simp add: con-def)
 moreover
 have (valor (interpretacion-asociada T)(No F)) = Verdad using hip
       by (simp add: v-negacion-def)
 thus ?thesis using hip by (simp add: con-def)
qed
```

Lema 4.11 *Sea T un conjunto de índices de símbolos proposicionales. Supongamos que $F = p_n$ y $n \in T$. Entonces $v'_T(F) = \mathsf{V}$.*

```
lemma valor-atomica-pertenece:
 assumes F = Atom k and k ∈ T
 shows valor (interpretacion-asociada T) F = Verdad
using assms
by (simp-all add: interpretacion-asociada-def)
```

Lema 4.12 *Sea T un conjunto de índices de símbolos proposicionales. Supongamos que $F = p_n$ y $n \notin T$. Entonces $v'_T(F) = \mathsf{F}$.*

```
lemma valor-atomica-no-pertenece:
 assumes F = Atom k and k ∉ T
 shows valor (interpretacion-asociada T) F = Falso
using assms
by (simp-all add: interpretacion-asociada-def)
```

Lema 4.13 *Si F es p_k y $k \in T$, entonces $hip_T(F) = \{p_k\}$.*

```
lemma hip-atomica-pertenece:
 assumes F = Atom k and k ∈ T
 shows hip T F = {Atom k}
using assms valor-atomica-pertenece
by simp-all
```

Lema 4.14 *Si F es p_k y $k \notin T$, entonces $hip_T(F) = \{\neg p_k\}$.*

lemma *hip-atomica-no-pertenece*:
assumes $F = Atom\ k$ **and** $k \notin T$
shows $hip\ T\ F = \{No\ (Atom\ k)\}$
using *assms valor-atomica-no-pertenece*
by *simp-all*

Lema 4.15 *Si F es* p_k *y* $k \in T$*, entonces* $con_T(F) = p_k$.

lemma *con-atomica-pertenece*:
assumes $F = Atom\ k$ **and** $k \in T$
shows $con\ T\ F = Atom\ k$
using *assms valor-atomica-pertenece con-def*
by *simp-all*

Lema 4.16 *Si F es* p_k *y* $k \in T$*, entonces* $con_T(F) = \neg p_k$.

lemma *con-atomica-no-pertenece*:
assumes $F = Atom\ k$ **and** $k \notin T$
shows $con\ T\ F = No\ (Atom\ k)$
using *assms valor-atomica-no-pertenece con-def*
by *simp-all*

Con las definiciones anteriores, el lema de Kalmar puede enunciarse de la siguiente manera.

Lema 4.17 (Kalmar) *Sea F una fórmula y T un conjunto de índices de símbolos proposicionales, entonces* $hip_T(F) \vdash con_T(F)$.

lemma *kalmar*[*rule-format*]: $\forall F.\ size\ F = n \longrightarrow hips\ T\ F \vdash conclus\ T\ F$

Para la demostración usamos los siguientes lemas que dan condiciones suficientes para deducir, a partir de un conjunto cualquiera S, las fórmulas

$\neg(\neg F)$, $F \to G$, $\neg(F \to G)$.

Lema 4.18 (casoa) *Si* $S \vdash G$ *entonces* $S \vdash F \to G$.

Demostración:
1 $S \vdash G \to (F \to G)$ [lema 1.13]
2 $S \vdash F \to G$ [MP, 1 y hipótesis]

□

lemma *casoa*:
assumes $S \vdash G$
shows $S \vdash F \to G$
proof −
have $S \vdash G \to (F \to G)$ **by** (*rule Axioma1*)
thus *?thesis* **using** *assms* **by** (*rule mp*)
qed

Lema 4.19 (casob) *Si* $S \vdash F$ *y* $S \vdash \neg G$ *entonces* $S \vdash \neg(F \to G)$.

Demostración:

1	$S \vdash F \to (\neg G \to \neg(F \to G))$	[lema 1.32]
2	$S \vdash \neg G \to \neg(F \to G)$	[MP, 1 y hipótesis]
3	$S \vdash \neg(F \to G)$	[MP, 2 y hipótesis]

□

lemma *casob*:
assumes $S \vdash F$
and $S \vdash No\ G$
shows $S \vdash No\ (F \to G)$
proof −
have $S \vdash F \to (No\ G \to No\ (F \to G))$ **by** (*rule negacion-del-condicional*)
hence $S \vdash No\ G \to No\ (F \to G)$ **using** *assms*(1) **by** (*rule mp*)
thus *?thesis* **using** *assms*(2) **by** (*rule mp*)
qed

Lema 4.20 (casoc) *Si* $S \vdash \neg F$ *entonces* $S \vdash F \to G$.

Demostración:

1	$S \vdash \neg F \to (F \to G)$	[lema 1.29]
2	$S \vdash F \to G$	[MP, 1 y hipótesis]

□

lemma *casoc*:
assumes $S \vdash No\ F$
shows $S \vdash F \to G$
proof −
have $S \vdash (No\ F \to (F \to G))$ **by** (*rule elimNeg*)
thus *?thesis* **using** *assms* **by** (*rule mp*)
qed

Los siguientes dos lemas servirán para simplificar la longitud de la prueba del lema de Kalmar.

Lema 4.21 *Si F es $\neg G$, entonces, para todo conjunto de fórmulas S y todo conjunto de índices T se tiene que*

$$S \vdash con_T(G) \to con_T(F).$$

Demostración: Por casos según el valor de $v_T(G)$.

Caso 1: Supongamos que $v_T(G) = \mathsf{V}$. Entonces, $con_T(G) = G$ y $v_T(F) = \mathsf{F}$. Luego, $con_T(F) = \neg F = \neg\neg G$. Por tanto, usando el lema 1.28, se tiene que
$S \vdash con_T(G) \to con_T(F)$.

Caso 2: Supongamos que $v_T(G) = \mathsf{F}$. Entonces, $con_T(G) = \neg G$ y $v_T(F) = \mathsf{V}$. Luego, $con_T(F) = F = \neg G$. Por tanto, usando el lema 1.18, se tiene que
$S \vdash con_T(G) \to con_T(F)$.

□

```
lemma con-negacion:
 assumes F = No G
 shows S ⊢ con T G → con T F
proof (cases valor (interpretacion-asociada T) G)
 — Caso 1
 { assume hip: (valor (interpretacion-asociada T) G) = Verdad
  hence 1: con T G = G by (simp add: con-def)
  have (valor (interpretacion-asociada T) F) = Falso using assms hip
        by (simp add: v-negacion-def)
  hence con T F = No F by (simp add: con-def)
  hence con T F = No No G using assms by simp
  thus S ⊢ con T G → con T F
   using 1 by (simp add: introDN) }
next
 — Caso 2
 { assume hip: valor (interpretacion-asociada T) G = Falso
  hence 1: con T G = No G by (simp add: con-def)
  have valor (interpretacion-asociada T) F = Verdad using assms hip
            by(simp add: v-negacion-def)
  hence con T F = F by (simp add: con-def)
  hence con T F = No G using assms by simp
  thus S ⊢ con T G → con T F
   using 1 by (simp add: identidad) }
qed
```

Lema 4.22 *Si F es $G \to H$, $S \vdash con_T(G)$ y $S \vdash con_T(H)$. Entonces, $S \vdash con_T(F)$.*

Demostración: Por casos según $v_T(H)$.

Caso 1: Supongamos que $v_T(H) = \mathsf{V}$, entonces $con_T(H) = H$, $v_T(F) = \mathsf{V}$ y $con_T(F) = G \to H$. Luego, por el lema 4.18, $S \vdash con_T(F)$.

Caso 2: Supongamos que $v_T(H) = \mathsf{F}$. Lo demostramos por casos según $v_T(G)$.

Caso 2a: Supongamos que $v_T(G) = \mathsf{V}$. Entonces, $con_T(G) = G$, $con_T(H) = \neg H$, $con_T(F) = \neg F$. Luego, por el lema 4.19, $S \vdash con_T(F)$.

Caso 2b: Supongamos que $v_T(G) = \mathsf{F}$. Entonces, $con_T(G) = \neg G$, $con_T(F) = F$. Luego, por el lema 4.20, $S \vdash con_T(F)$.

□

```
lemma con-implicacion:
 assumes F = G → H
 and S ⊢ con T G
 and S ⊢ con T H
 shows S ⊢ con T F
proof (cases valor (interpretacion-asociada T) H)
 — Caso 1
 { assume hip: valor (interpretacion-asociada T) H = Verdad
  hence 1: con T H = H using con-def by simp
  have valor (interpretacion-asociada T) F = Verdad using assms(1) hip
         by (simp add: v-implicacion-def)
  hence con T F = G → H using con-def assms(1) by simp
  thus S ⊢ con T F using casoa 1 assms(3) by simp }
next
 — Caso 2
 { assume hip: valor (interpretacion-asociada T) H = Falso
  show S ⊢ con T F
  proof (cases valor (interpretacion-asociada T) G)
  — Caso 2a
  { assume hip1: valor (interpretacion-asociada T) G = Verdad
   hence 1: con T G = G using con-def by simp
   have 2: con T H = No H using hip con-def by simp
   hence valor (interpretacion-asociada T) F = Falso using hip hip1 assms(1)
          by(simp add: v-implicacion-def)
   hence con T F = No F by (simp add: con-def)
   thus S ⊢ con T F using casob 1 2 assms by simp }
  next
  — Caso 2b
  { assume hip: valor (interpretacion-asociada T) G = Falso
   hence 1: con T G = No G using con-def by simp
   have valor (interpretacion-asociada T) F = Verdad using hip assms(1)
         by(simp add: v-implicacion-def)
   hence con T F = F using con-def by simp
```

thus $S \vdash$ *con T F* **using** *casoc 1 assms* **by** *simp* }
qed }
qed

Lema 4.23 (Kalmar) *Sea F una fórmula y T un conjunto de índices de símbolos proposicionales, entonces* $hip_T(F) \vdash con_T(F)$.

Demostración: Por inducción en el tamaño de la fórmula. Sea n un número natural. La hipótesis de inducción (HI) afirma que para toda fórmula F de tamaño menor que n se tiene $hip_T(F) \vdash con_T(F)$. Tenemos que demostrar que para toda fórmula F de tamaño n se tiene $hip_T(F) \vdash con_T(F)$.

Sea F una fórmula. Supongamos (HI1) que el tamaño de F es n, Tenemos que demostrar que $hip_T(F) \vdash con_T(F)$. Lo haremos por distinción de casos según la forma de F.

Caso 1: Supongamos que F es la fórmula atómica p_k. Distinguimos dos casos:

Caso 1a: Supongamos que $k \in T$. Entonces
1. $hip_T(F) = \{p_k\}$ [Por 4.13]
2. $con_T(F) = p_k$ [Por 4.15]
3. $hip_T(F) \vdash con_T(F)$ [Por 1, 2 y 1.10]

Caso 1b: Supongamos que $k \notin T$. Entonces
1. $hip_T(F) = \{\neg p_k\}$ [Por 4.14]
2. $con_T(F) = \neg p_k$ [Por 4.16]
3. $hip_T(F) \vdash con_T(F)$ [Por 1, 2 y 1.10]

Caso 2: Supongamos que F es $\neg G$. Entonces,
1. $hip_T(G) \vdash con_T(G)$ [Por H.I.]
2. $hip_T(F) = hip_T(G)$ [Por 4.4]
3. $hip_T(F) \vdash con_T(G)$ [Por 1 y 2]
4. $hip_T(F) \vdash con_T(G) \to con_T(F)$ [Por 4.21]
5. $hip_T(F) \vdash con_T(F)$ [Por MP, 4 y 3]

Caso 3: Supongamos que F es $G \to H$. Entonces, se tiene que
$hip_T(F) \vdash con_T(G)$ (1)
En efecto, la longitud de G es menor que la longitud de F y, por (HI1) es menor que n. Luego, por (HI), se tiene $hip_T(G) \vdash con_T(G)$. Además, $hip_T(G) \subseteq hip_T(F)$. Por lo tanto, usando la regla de debilitamiento, se tiene que $hip_T(F) \vdash con_T(G)$.

De la misma forma, se tiene que,
$hip_T(F) \vdash con_T(H)$ (2)

Por el lema 4.22, junto con (1) y (2), se tiene que $hip_T(F) \vdash con_T(F)$.

□

```
lemma kalmar[rule-format]:
 ∀ F. size F = n ⟶ hip T F ⊢ con T F
proof (induct rule: nat-less-induct)
 fix n
 assume HI: ∀ m<n. ∀ F. size F = m ⟶ hip T F ⊢ con T F
 show  ∀ F. size F = n ⟶ hip T F ⊢ con T F
 proof
  fix  F::formula
  show size F = n ⟶ hip T F ⊢ con T F
  proof
   assume HI1: size F = n
   show hip T F ⊢ con T F
   proof (cases F)
   — Caso 1:
   { fix k
    assume Caso1: F = Atom k
    show hip T F ⊢ con T F
    proof (cases k ∈ T)
     — Caso 1a
     { assume Caso1a: k ∈ T
      hence hip T F = {Atom k}
       using Caso1 hip-atomica-pertenece by simp
      moreover hence con T F = Atom k
       using Caso1 Caso1a con-atomica-pertenece by simp
      ultimately show hip T F ⊢ con T F using hip by simp }
     — Caso 1b
     { assume Caso1a: k ∉ T
      hence hip T F = {No (Atom k)}
       using Caso1 hip-atomica-no-pertenece by simp
      moreover hence con T F = No (Atom k)
       using Caso1 Caso1a con-atomica-no-pertenece by simp
      ultimately show hip T F ⊢ con T F using hip by simp }
    qed }
    next
    — Caso 2:
    { fix G
     assume Caso2: F = No G
     hence size G < size F by simp
     hence b1: hip T G ⊢ con T G using HI and HI1 by simp
     have b2: hip T F = hip T G using Caso2 by simp
     hence b3: hip T F ⊢ con T G using b1 by simp
     have b4: hip T F ⊢ con T G ⟶ con T F
      using Caso2 con-negacion by simp
```

```
      show hip T F ⊢ con T F
        using b4 b3 by (rule mp) }
    next
    — Caso 3:
    { fix G H
      assume Caso3: F = G → H
      have 1: hip T F ⊢ con T G
      proof −
        have size G < size F using Caso3 by simp
        hence hip T G ⊢ con T G using HI and HI1 by simp
        moreover
        have hip T G ⊆ hip T F using Caso3 by auto
        ultimately
        show hip T F ⊢ con T G by (rule debilitamiento)
      qed
      have 2: hip T F ⊢ con T H
      proof −
        have size H < size F using Caso3 by simp
        hence hip T H ⊢ con T H using HI and HI1 by simp
        moreover
        have hip T H ⊆ hip T F using Caso3 by auto
        ultimately
        show hip T F ⊢ con T H by (rule debilitamiento)
      qed
    show hip T F ⊢ con T F
      using con-implicacion 1 2 Caso3 by simp }
  qed
 qed
qed
qed
```

El siguiente lema de eliminación de variables permitirá la demostración del teorema de completitud

Lema 4.24 (Lema de eliminación de variables) *Sea F una tautología. Si H es un conjunto finito, entonces para todo conjunto de índices T se tiene que, $hip_T(F) - H \vdash F$.*

Demostración: Por inducción finita sobre H.

Base: Supongamos que $H = \emptyset$. Sea T un conjunto de índices. Se tiene

$$\begin{aligned} hip_T(F) - H &= hip_T(F) - \emptyset \\ &= hip_T(F) \\ &\vdash con_T(F) && \text{[por el lema 4.23]} \\ &= F && \text{[por ser } F \text{ tautología]} \end{aligned}$$

Paso de inducción Sea H un conjunto finito y $x \notin H$. La hipótesis de inducción es

$$\forall T.(hip_T(F) - H \vdash F \qquad \text{(HI)}$$

Hay que probar que

$$\forall T.(hip_T(F) - (\{x\} \cup H) \vdash F$$

Sea T un conjunto de índices. Demostraremos que

$$hip_T(F) - (\{x\} \cup H) \vdash F \qquad (1)$$

distinguiendo dos casos.

Caso 1: Supongamos que $x \in hip_T(F)$. Entonces, por el lema4.6, existe un k tal que x es p_k con $k \in T$ ó x es $\neg p_k$ con $k \notin T$. Demostraremos (1) distinguiendo dos subcasos:

Caso 1a: Supongamos que x es p_k con $k \in T$. Por el lema 1.35, basta probar que

$$\{p_k\} \cup (hip_T(F) - (\{x\} \cup H)) \vdash F \qquad \text{(1a1)}$$
$$\{\neg p_k\} \cup (hip_T(F) - (\{x\} \cup H)) \vdash F \qquad \text{(1a2)}$$

La demostración de (1a1) es la siguiente:

1. $hip_T(F) - H \subseteq \{p_k\} \cup (hip_T(F) - (\{x\} \cup H))$ [porque x es p_k]
2. $hip_T(F) - H \vdash F$ [por la HI]
3. $\{p_k\} \cup (hip_T(F) - (\{x\} \cup H)) \vdash F$ [por 1, 2 y el lema 1.11]

La demostración de (1a2) es la siguiente:

1. $hip_{T-\{k\}}(F) \subseteq (\neg p_k) \cup (hip_T(F) - \{p_k\})$ [por el lema 4.7]
2. $hip_{T-\{k\}}(F) - H \subseteq (\neg p_k) \cup (hip_T(F) - (\{x\} \cup H))$ [por 1 y porque x es p_k]
3. $hip_{T-\{k\}}(F) - H \vdash F$ [por la HI]
4. $\{\neg p_k\} \cup (hip_T(F) - (\{x\} \cup H)) \vdash F$ [por 2, 3 y el lema 1.11]

Caso 1b: Supongamos que x es $\neg p_k$ con $k \notin T$. Por el lema 1.35, basta probar que

$$\{p_k\} \cup (hip_T(F) - (\{x\} \cup H)) \vdash F \qquad \text{(1b1)}$$
$$\{\neg p_k\} \cup (hip_T(F) - (\{x\} \cup H)) \vdash F \qquad \text{(1b2)}$$

La demostración de (1b1) es la siguiente:

1. $hip_{k \cup T}(F) \subseteq \{p_k\} \cup (hip_T(F) - \{\neg p_k\}$ por el lema 4.8
2. $hip_{k \cup T}(F) - H \subseteq \{p_k\} \cup (hip_T(F) - (\{x\} \cup H))$ [por 1 y porque x es $\neg p_k$]
3. $hip_{k \cup T}(F) - H \vdash F$ [por la HI]
4. $\{p_k\} \cup (hip_T(F) - (\{x\} \cup H)) \vdash F$ [por 2, 3 y el lema 1.11]

La demostración de (1b2) es la siguiente:

1. $hip_T(F) - H \subseteq \{\neg p_k\} \cup (hip_T(F) - (\{x\} \cup H))$ [porque x es $\neg p_k$]
2. $hip_T(F) - H \vdash F$ [por la HI]
3. $\{\neg p_k\} \cup (hip_T(F) - (\{x\} \cup H)) \vdash F$ [por 1, 2 y el lema 1.11]

Caso 2: Supongamos que $x \notin hip_T(F)$. Entonces,

1. $hip_T(F) - (\{x\} \cup H) = hip_T(F) - H$
2. $hip_T(F) - H \vdash F$ [por la HI]
3. $hip_T(F) - (\{x\} \cup H) \vdash F$ [por 1, 2]

□

lemma *eliminacion-variables*:
assumes *tautologia F*
shows *finite H* $\Longrightarrow \forall T.\ ((hip\ T\ F) - H) \vdash F$
proof (*induct rule*: *finite-induct*)
— Base:
{ **fix** *T*
have $(hip\ T\ F) - \{\} = hip\ T\ F$ **by** *simp*
moreover have $hip\ T\ F \vdash con\ T\ F$ **using** *kalmar* **by** *simp*
moreover have $con\ T\ F = F$
using *assms con-def tautologia-def* **by** *simp*
ultimately have $(hip\ T\ F) - \{\} \vdash F$ **by** *simp* }
thus $\forall T.\ ((hip\ T\ F) - \{\}) \vdash F$ **by** *simp* — Fin de la base
next
— Paso:
{ **fix** *x H* **assume**
finite H **and**
$x \notin H$ **and**
HI: $\forall T.\ ((hip\ T\ F) - H) \vdash F$
show $\forall T.\ ((hip\ T\ F) - (insert\ x\ H)) \vdash F$
proof
{ **fix** *T*
show $(hip\ T\ F) - (insert\ x\ H) \vdash F$
proof (*cases* $x \in (hip\ T\ F)$)
— Caso 1:
{ **assume** *1*: $x \in (hip\ T\ F)$
obtain *k* **where** *2*: $(x = Atom\ k \wedge k \in T) \vee (x = No\ (Atom\ k) \wedge k \notin T)$
using *1 pertenece-hip* **by** *blast*
show $(hip\ T\ F) - (insert\ x\ H) \vdash F$
proof (*cases* $x = Atom\ k \wedge k \in T$)
— Caso 1a:
{ **assume** $x = Atom\ k \wedge k \in T$
show $(hip\ T\ F) - (insert\ x\ H) \vdash F$
proof −
— Caso 1a1:
have $\{Atom\ k\} \cup ((hip\ T\ F) - (insert\ x\ H)) \vdash F$
proof −
have $(hip\ T\ F) - H \subseteq$
$insert\ (Atom\ k)\ ((hip\ T\ F) - (insert\ (Atom\ k)\ H))$ **by** *auto*
moreover
from *HI* **have** $(hip\ T\ F) - H \vdash F$ **by** *simp*
ultimately
have $insert\ (Atom\ k)\ ((hip\ T\ F) - (insert\ (Atom\ k)\ H)) \vdash F$
using *debilitamiento* **by** *simp*

```
    thus {Atom k} ∪ ((hip T F) − (insert x H)) ⊢ F
      using ⟨x = Atom k ∧ k ∈ T⟩ by simp
   qed — Fin del Caso 1a1
   moreover
   — Caso 1a2:
   have {No (Atom k)} ∪ ((hip T F) − (insert x H)) ⊢ F
   proof −
    have hip (T−{k}) F ⊆
         insert (No (Atom k)) ((hip T F) − {Atom k})
      using hip-diferencia by simp
    hence (hip (T−{k}) F) − H ⊆
         {No (Atom k)} ∪ ((hip T F) − (insert x H))
      using ⟨x = Atom k ∧ k ∈ T⟩ by auto
    moreover
    from HI have (hip (T−{k}) F) − H ⊢ F by simp
    ultimately
    show {No (Atom k)} ∪ ((hip T F) − (insert x H)) ⊢ F
      using debilitamiento by simp
   qed — Fin del Caso 1a2
   ultimately show (hip T F) − (insert x H) ⊢ F
    using tercio-excluso-2[of Atom k (hip T F)−(insert x H)]
    by simp
  qed } — Fin del Caso 1a
next
— Caso 1b:
{ assume ¬(x = Atom k ∧ k ∈ T)
  hence x = No (Atom k) ∧ k ∉ T
   using 2 by blast
  show (hip T F) − (insert x H) ⊢ F
  proof −
   — Caso 1b1:
   have {Atom k} ∪ ((hip T F) − (insert x H)) ⊢ F
   proof −
    have hip ({k}∪T) F ⊆ {Atom k}∪((hip T F)−{No (Atom k)})
      by (rule hip-cons)
    hence (hip ({k}∪ T) F) − H ⊆
         {Atom k}∪((hip T F) − ({x}∪H))
      using ⟨x = No (Atom k) ∧ k ∉ T⟩ by auto
    moreover
    from HI have (hip ({k}∪T) F) − H ⊢ F by simp
    ultimately
    show {Atom k} ∪ ((hip T F) − (insert x H)) ⊢ F
      using debilitamiento by simp
```

```
          qed — Fin del Caso 1b1
          moreover
          — Caso 1b2:
          have {No (Atom k)} ∪ ((hip T F) − (insert x H)) ⊢ F
          proof −
           have (hip T F) − H ⊆
             {No (Atom k)} ∪ ((hip T F) − (insert x H))
             using ⟨x = No (Atom k) ∧ k ∉ T⟩ by auto
           moreover
           from HI have (hip T F) − H ⊢ F by simp
           ultimately
           show {No (Atom k)} ∪ ((hip T F) − ((insert x H))) ⊢ F
            using debilitamiento by simp
          qed — Fin del Caso 1b2
          ultimately
          show (hip T F) − (insert x H) ⊢ F
           using tercio-excluso-2[of Atom k (hip T F)−(insert x H)]
           by simp
         qed } — Fin del Caso 1b
       qed } — Fin del Caso 1
      next
      — Caso 2
      { assume x ∉ (hip T F)
       show (hip T F) − (insert x H) ⊢ F
       proof −
        have (hip T F) − (insert x H) = (hip T F) − H
         using ⟨x ∉ (hip T F)⟩ by simp
        moreover
        have (hip T F) − H ⊢ F using HI by simp
        ultimately show (hip T F) − (insert x H) ⊢ F by simp
       qed } — Fin del Caso 2
     qed }
  qed } — Fin del paso
qed
```

Teorema 4.25 (de completitud ([1] p. 42)) *Todas las tautologías son deducibles, es decir, para toda fórmula F, si $\models F$, entonces $\vdash F$.*

Demostración: Sea F una tautología. Por el lema 4.5, $hip_T(F)$ es un conjunto finito y, por el lema 4.24, $hip_T(F) - hip_T(F) \vdash F$. Por tanto, $\emptyset \vdash F$.

□

theorem *completitud*:

assumes *tautologia F*
shows $\{\} \vdash F$
proof $-$
have *finite* (*hip T F*) **by** (*rule hip-finito*)
hence (*hip T F*) $-$ (*hip T F*) $\vdash F$
using *assms eliminacion-variables* **by** *blast*
thus $\{\} \vdash F$ **by** *simp*
qed

- $S \vdash F \to (G \to F)$ (*Axioma1*)
- $S \vdash F \to (G \to H) \to (F \to G \to (F \to H))$ (*Axioma2*)
- $S \vdash No\ G \to No\ F \to (No\ G \to F \to G)$ (*Axioma3*)
- $\dfrac{F \in S}{S \vdash G \to F}$ (*por-hip*)
- $\dfrac{F \to G \in S \qquad F \in S}{S \vdash G}$ (*mp-con-hip*)
- $S \vdash F \to F$ (*identidad*)
- $\dfrac{S \vdash F \qquad S \subseteq T}{T \vdash F}$ (*debilitamiento*)
- $\dfrac{S \vdash F}{S \cup \{G\} \vdash F}$ (*debilitamiento2*)
- $\dfrac{S \vdash F}{\forall\, G\ S'.\ S = \{G\} \cup S' \longrightarrow S' \vdash G \to F}$ (*lema-deduccion*)
- $\dfrac{\{F\} \cup S \vdash G}{S \vdash F \to G}$ (*deduccion*)
- $\dfrac{S \vdash F \qquad \{F\} \cup S \vdash G}{S \vdash G}$ (*regla-de-corte*)
- $\{F \to G, G \to H\} \vdash F \to H$ (*encadenamiento*)
- $\dfrac{S \vdash F \to G \qquad S \vdash G \to H}{S \vdash F \to H}$ (*encadenamiento2*)
- $\{F \to (G \to H), G\} \vdash F \to H$ (*mp2*)
- $\dfrac{S \vdash F \to (G \to H) \qquad S \vdash G}{S \vdash F \to H}$ (*mp2a*)
- $\dfrac{S \vdash F \qquad S \vdash G \qquad \{F, G\} \vdash H}{S \vdash H}$ (*corte2*)
- $S \vdash No\ No\ F \to F$ (*eliminaDN*)
- $S \vdash F \to No\ No\ F$ (*introDN*)

- $S \vdash No\ F \to (F \to G)$ (*elimNeg*)
- $S \vdash No\ G \to No\ F \to (F \to G)$ (*contrareciproco-1*)
- $S \vdash F \to G \to (No\ G \to No\ F)$ (*contrareciproco-2*)
- $S \vdash F \to (No\ G \to No\ (F \to G))$ (*negacion-del-condicional*)
- $S \vdash F \to G \to (No\ F \to G \to G)$ (*tercio-excluso-1*)
- $$\frac{\{F\} \cup S \vdash G \qquad \{No\ F\} \cup S \vdash G}{S \vdash G}$$ (*tercio-excluso*)
- $$\frac{\{F\} \cup S \vdash G \qquad \{No\ F\} \cup S \vdash G}{S \vdash G}$$ (*tercio-excluso-2*)
- *tautologia* $(F \to (G \to F))$ (*tautologia-A1*)
- *tautologia* $(F \to (G \to H) \to (F \to G \to (F \to H)))$ (*tautologia-A2*)
- *tautologia* $(No\ G \to No\ F \to (No\ G \to F \to G))$ (*tautologia-A3*)
- $\forall\ I.\ I\ mod\ \emptyset$ (*modelo-de-vacio*)
- *tautologia* $F = \emptyset \models F$ (*CNS-tautologia*)
- $$\frac{tautologia\ F}{S \models F}$$ (*CNS-tautologia1*)
- $$\frac{S \vdash F}{S \models F}$$ (*validez*)
- *finite* $(hip\ T\ F)$ (*hip-finito*)
- $$\frac{G \in hip\ T\ F}{\exists\ n.\ G = Atom\ n \wedge n \in T \vee G = No\ Atom\ n \wedge n \notin T}$$ (*pertenece-hip*)
- $hip\ (T - \{n\})\ F \subseteq \{No\ Atom\ n\} \cup (hip\ T\ F - \{Atom\ n\})$ (*hip-diferencia*)
- $hip\ (\{n\} \cup T)\ F \subseteq \{Atom\ n\} \cup (hip\ T\ F - \{No\ Atom\ n\})$ (*hip-cons*)
- $$\frac{F = Atom\ k \qquad k \in T}{valor\ (interpretacion\text{-}asociada\ T)\ F = Verdad}$$ (*valor-atomica-pertenece*)
- $$\frac{F = Atom\ k \qquad k \notin T}{valor\ (interpretacion\text{-}asociada\ T)\ F = Falso}$$ (*valor-atomica-no-pertenece*)

- $\dfrac{F = Atom\ k \qquad k \in T}{hip\ T\ F = \{Atom\ k\}}$ (*hip-atomica-pertenece*)
- $\dfrac{F = Atom\ k \qquad k \notin T}{hip\ T\ F = \{No\ Atom\ k\}}$ (*hip-atomica-no-pertenece*)
- $\dfrac{F = Atom\ k \qquad k \in T}{con\ T\ F = Atom\ k}$ (*con-atomica-pertenece*)
- $\dfrac{F = Atom\ k \qquad k \notin T}{con\ T\ F = No\ Atom\ k}$ (*con-atomica-no-pertenece*)
- $\dfrac{S \vdash G}{S \vdash F \rightarrow G}$ (*casoa*)
- $\dfrac{S \vdash F \qquad S \vdash No\ G}{S \vdash No\ (F \rightarrow G)}$ (*casob*)
- $\dfrac{S \vdash No\ F}{S \vdash F \rightarrow G}$ (*casoc*)
- $\dfrac{F = No\ G}{S \vdash con\ T\ G \rightarrow con\ T\ F}$ (*con-negacion*)
- $\dfrac{F = G \rightarrow H \qquad S \vdash con\ T\ G \qquad S \vdash con\ T\ H}{S \vdash con\ T\ F}$ (*con-implicacion*)
- $\dfrac{size\ F = n}{hip\ T\ F \vdash con\ T\ F}$ (*kalmar*)
- $\dfrac{tautologia\ F \qquad finite\ H}{\forall\, T.\ hip\ T\ F - H \vdash F}$ (*eliminacion-variables*)
- $\dfrac{tautologia\ F}{\emptyset \vdash F}$ (*completitud*)


Kalmar
• debilitamiento
• hip-atomica-pertenece : valor-atomica-pertenece
• con-atomica-pertenece : valor-atomica-pertenece
• hip-atomica-no-pertenece : valor-atomica-no-pertenece
• con-atomica-no-pertenece : valor-atomica-no-pertenece
• con-negac
identidad
Axioma1
Axioma2
introDN
Axioma1
Axioma3
encad2
debilita2
deduccion : lema-deducc
Axioma1
Axioma2
identidad
por-hip : Axioma1
eliminaDN
Axioma1
Axioma3
encad2
identidad
mp2a
debilita2
deduccion
• con-implicacion
casoa : Axioma1
casob : negac-condic
deduccion
encad2
contrarec2
deduccion
encad2
eliminaDN
contrarec1
Axioma1
Axioma3
deduccion
encad2
casoc : elimNeg
Axioma1
Axioma3
deduccion



Referencias

[1] E. Mendelson. *Introduction to mathematical logic*. CRC Press, 2001.